

40 PROGRAMMES POUR CASIO PB-700

INTRODUCTION

L'auteur remercie pour leur participation à l'ouvrage :

M. Alexandre OCANA (société NOBLET)
M. André TONIC
M. Bertrand RAVEL

Les programmes de cet ouvrage sont disponibles sur cassette. Dupliquée et distribuée par D.D.I., elle est en vente dans les boutiques d'informatique et les librairies techniques.

Vente par correspondance :
L.P.R., 43, rue de Dunkerque, 75010 Paris,
Tél. : 878.09.92.

Les quarante programmes de ce recueil ont été conçus pour illustrer, par des applications utiles ou amusantes, quelques-unes des nombreuses fonctions du Basic Casio ; ainsi se trouve, à la suite de chaque programme, un index des principales fonctions utilisées. J'ai souhaité par ailleurs que ces exemples d'application soient pour le lecteur générateur d'améliorations, d'extensions de nouveaux programmes. En un mot, qu'ils soient le point de départ d'une meilleure compréhension et d'une meilleure utilisation des puissantes possibilités de cette machine, et plus particulièrement de son imprimante traçante d'une puissance tout à fait originale.

Outre l'aspect didactique ou créatif, il faut garder à l'esprit que chacun des programmes est immédiatement utilisable ; l'exemple ou la figure l'accompagnant permettra une vérification rapide de son bon fonctionnement.

1) Tous les programmes sont opérationnels avec le PB 700 en version de base, seuls les programmes de « *Graphisme* » nécessitent le raccordement de l'imprimante. Avant toute saisie, je vous recommande de vous assurer que la zone de programme où il va être introduit est bien disponible. Dans le cas contraire, procéder à NEW ENTER. Ensuite, le programme sera saisi conformément aux spécifications du manuel d'utilisation de la machine. Pour faciliter le repérage, la présentation des programmes a été standardisée en les numérotant de dix en dix. Seules les remarques (REM) possèdent des numéros de

ligne finissant par cinq de façon à pouvoir les distinguer du reste du programme, soit pour l'étude, soit pour leur suppression.

2) Utilisation des programmes. Une fois en machine, le programme est mis en route par RUN RETURN ou SHIFT PN avec N correspondant à la zone où il se trouve. Le détail de l'utilisation de chaque programme est donné au chapitre mode d'emploi et les manipulations sont décrites dans leur ordre exact. Les réponses de l'utilisateur pouvant être de forme constante ou variable, chaque type est regroupé sous un mode d'entrées propre, soit : les constantes par INKEY\$ et les variables par INPUT. Dans les deux cas, des parenthèses le signalent dans le détail de l'utilisation.

Enfin, pour ce qui concerne les modifications pouvant être apportées, je suis certain qu'elles ne manqueront pas et que l'imagination du lecteur s'envolera au contact d'une si belle machine.

G.P.

SOMMAIRE

JEUX

Tirage de cartes.....	10
Tirage de dés.....	12
Jack-pot.....	15
Pendu	19
Carnival	23
Super mind.....	25
Allumettes	29
Logic 100.....	33
Grapho mémoire.....	36
Divination.....	40
Caverns of Thora.....	43

GRAPHISME

Cercles colorés.....	46
Ellipses colorées.....	49
Télécran	52

Graphismes divers.....	54
Triangles divers.....	57
Carrés divers.....	60
Inversion vidéo.....	63

MATHEMATIQUES

Conversion arabes-romains.....	65
Conversion de températures.....	68
Volumes divers.....	71
Polaire-rectangulaire	74
Nombres premiers.....	77
Fractions	80
PGCD-PPCM	83
Grandes factorielles.....	85

VIE PRATIQUE

Signalisation phonétique.....	87
Morse	90
Dates.....	94
Agenda.....	98
Conversion julien-grégorien.....	102
Conversion acres-hectares.....	105
Poids idéal.....	107
Espérance de vie.....	109
Monnaie	112
Surface corporelle.....	115

UTILITAIRES

Copie d'écran.....	117
Dactylographie	119
Machine à écrire.....	122
Titres.....	125

TIRAGE DE CARTES

PRESENTATION

Indispensable à la réalisation de jeux plus évolués, tels que le Poker, le Black-Jack ou même la Belote, le tirage de cartes doit générer des mains de 1 à 5 cartes prises dans un jeu de 52 cartes, et cela sans répétition.

MODE D'EMPLOI

- 1) Sélectionner le nombre de cartes voulues (en 1 et 5 par INKEY\$).
- 2) Affichage du tirage avec « 0 » valant 10 (presser « * » pour continuer, INKEY\$).

FONCTIONS

ASC, CHR\$, CLS, DRAW, INKEY\$, INT, LOCATE, MID\$, PRINT TAB, RND, VAL.

PROGRAMME

```

5 REM *** TIRAGE DE CARTES ***
10 CLS :CLEAR :BEEP
20 PRINT "***** TIRAGE ***** DE 1 A
5 CARTES ?"
30 A$=INKEY$:B=VAL(A$):IF B>0 THEN IF
B<6 THEN 50


```

```

40 BEEP :GOTO 30
45 REM *TIRAGE DES CARTES*
50 CLS :FOR J=0 TO B-1:C$=CHR$(INT(RN
D*4)+232)
60 U$=MID$("OUDRA",INT(RND*5)+1,1):IF
RND<.8 THEN U$=STR$(INT(RND*8)+2)
65 REM *AFFICHAGE*
70 LOCATE J+3*K,0:PRINT C$;" ";C$:LOC
ATE J+3*K,2:PRINT C$;" ";C$
80 LOCATE J+3*K,1:IF ASC(U$)>57 THEN
LOCATE J+3*K+1,1
90 PRINT U$":BEEP :NEXT J
100 FOR G=1 TO B:DRAW(28+(G-1)*32,0)-(28+(G-1)*32,28):NEXT G
110 DRAW(0,28)-(28+(B-1)*32,28)
120 IF INKEY$<>"*" THEN 120
130 CLS :BEEP 1:PRINT TAB(20);"UN AUTR
E TIRAGE:O/N?"
140 IF INKEY$="O" THEN 10
150 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR K=1 TO 300:NEXT K:END
160 GOTO 140

```

EXEMPLE



TIRAGE DE DES

PRESENTATION

Exemple de graphisme très réaliste, ce tirage de dés pourra intervenir dans nombre de programmes de jeux de dés tels que le 421, le Notouane ou encore le Yams.

MODE D'EMPLOI

- 1) Choisir le nombre de dés devant apparaître (de 1 à 7 pour INKEY\$).
- 2) Affichage de l'ensemble du tirage (presser « * » pour continuer, INKEY\$).
- 3) Pour un autre tirage, presser « O » sinon « N » (INKEY\$).

FONCTIONS

CLS, DRAW, INKEY\$, INT, PRINT TAB, RND, VAL.

PROGRAMME

```

5 REM *** TIRAGE DE DES ***
10 CLS :CLEAR
20 PRINT "***** TIRAGE ***** DE 1
A 7 DES ?"
30 BEEP :A$=INKEY$:B=VAL(A$):IF B>0 T
HEN IF B<8 THEN 50

```

```

40 GOTO 30
45 REM *TIRAGES ET AFFICHAGE*
50 CLS :FOR J=0 TO B-1:BEEP :U=J*24:D
RAW(U,0)-(U+15,0):DRAW(U,15)-(U+15,15)
60 DRAW(U,0)-(U,15):DRAW(U+15,0)-(U+1
5,15)
70 T=INT(RND*6)+1:GOSUB T*10+120:NEXT
J
80 IF INKEY$<>"*" THEN 80
90 CLS :PRINT TAB(20);"UN AUTRE TIRAG
E:O/N?":BEEP 1
100 IF INKEY$="O" THEN 10
110 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR K=1 TO 300:NEXT K:END
120 GOTO 100
125 REM *ROUTINES GRAPHIQUES*
130 H=J*24+7:K=H+1:DRAW(H,7)-(K,7):DRA
W(H,8)-(K,8):RETURN
140 H=J*24+2:K=H+1:DRAW(H,2)-(K,2):DRA
W(H,3)-(K,3)
141 DRAW(H+10,12)-(H+11,12):DRAW(H+10,
13)-(H+11,13):RETURN
150 GOSUB 140:GOSUB 130:RETURN
160 GOSUB 140:H=J*24+12:K=H+1:Q=H-10:R
=Q+1
161 DRAW(H,2)-(K,2):DRAW(H,3)-(K,3)
162 DRAW(Q,12)-(R,12):DRAW(Q,13)-(R,13
):RETURN
170 GOSUB 160:GOSUB 130:RETURN
180 GOSUB 160:H=J*24+2:K=H+1:Q=J*24+12
:R=Q+1
190 DRAW(H,7)-(K,7):DRAW(H,8)-(K,8)
200 DRAW(Q,7)-(R,7):DRAW(Q,8)-(R,8):RE
TURN

```

EXEMPLE



JEUX

JACK-POT

PRESENTATION

A l'image des célèbres machines à sous, ce programme de Jack-Pot est une parfaite illustration ludico-informatique de l'utilisation des nombreux caractères disponibles sur le PB 700.

MODE D'EMPLOI

- 1) Introduire une mise inférieure au capital (1 000 au départ, INKEY\$).
- 2) Les cylindres tournent et s'arrêtent sur 3 signes choisis au hasard.
- 3) Les gains sont calculés sur la disposition des codes ASCII, soit :

CONFIGURATION	ASCII	GAINS
3 signes identiques	254	100 fois
3 signes identiques	qcq	75 fois
3 signes se suivant	qcq	50 fois
3 cartes	qcq	40 fois
3 chiffres	qcq	30 fois
3 majuscules	qcq	30 fois
3 minuscules	qcq	25 fois
2 signes identiques	qcq	15 fois
3 caractères katakana	qcq	10 fois
3 signes croissants	qcq	5 fois
3 signes décroissants	qcq	2 fois

FONCTIONS

ASC, CHR\$, CLS, INKEY\$, LOCATE, TAB, RND.

PROGRAMME

```
5 REM *** JACK-POT ***
7 CLEAR :DIM A$(2):C=1000
10 CLS :BEEP :LOCATE 0,1:PRINT "*****"
JACK-POT *****
15 REM *AFFICHAGE*
20 BEEP :LOCATE 0,2:PRINT " ";CHR$(28);";CHR$(228);CHR$(229);";CHR$(229);
25 PRINT CHR$(228);CHR$(229);";CHR$(229);
30 LOCATE 0,0:INPUT "VOTRE MISE ";M:I
F M>C THEN 10
40 LOCATE 0,0:PRINT "*****"
*****
50 FOR J=1 TO INT(RND*20+1):FOR K=0 T
O 2
60 B=INT(RND*223)+32:IF B>126 THEN IF
B<143 THEN 60
70 IF B>227 THEN IF B<232 THEN 60
80 A$(K)=CHR$(B):NEXT K
90 LOCATE 4,2:PRINT A$(0):LOCATE 9,2:
PRINT A$(1):LOCATE 14,2:PRINT A$(2):NEXT
J:BEEP 1
95 REM *POINTS*
100 X=ASC(A$(0)):Y=ASC(A$(1)):Z=ASC(A$
(2)):R=0
110 IF X=Y-1 THEN IF X=Z-2 THEN R=50
120 IF X=254 THEN IF YX THEN IF Z=Y TH
EN R=100
130 IF X=Y THEN IF X=Z THEN R=75
```

```
140 IF X>47 THEN IF X<58 THEN IF Y>47
THEN IF Y<58 THEN IF Z>47 THEN IF Z<58 T
HEN R=30
150 IF X>64 THEN IF X<91 THEN IF Y>64
THEN IF Y<91 THEN IF Z>64 THEN IF Z<91 T
HEN R=30
160 IF X>96 THEN IF X<123 THEN IF Y>96
THEN IF Y<123 THEN IF Z>96 THEN IF Z<12
3 THEN R=25
170 IF X>231 THEN IF X<236 THEN IF Y>2
31 THEN IF Y<236 THEN IF Z>231 THEN IF Z
<236 THEN R=40
180 IF X=Y THEN R=15
190 IF X=Z THEN R=15
200 IF Y=Z THEN R=15
210 IF X>165 THEN IF X<222 THEN IF Y>1
65 THEN IF Y<222 THEN IF Z>165 THEN IF Z
<222 THEN R=10
220 IF X>Y THEN IF Y>Z THEN R=5
230 IF X>Y THEN IF Y>Z THEN R=2
240 C=C-M+M*R:LOCATE 6,0:PRINT "MISE "
;CHR$(240);R
250 IF INKEY$<>"*" THEN 250
260 CLS :BEEP :PRINT "CAPITAL =";C;"$"
:IF C<1 THEN 310
270 PRINT "ON CONTINUE : O/N ?"
280 IF INKEY$="N" THEN PRINT TAB(20);"
AU REVOIR,A BIENTOT!":FOR F=1 TO 300:NEX
T F:END
290 IF INKEY$="O" THEN 10
300 GOTO 280
310 LOCATE 0,1:PRINT "**** RUINE !!!!"
****REJOUEZ-VOUS : O/N ?"
320 IF INKEY$="O" THEN 2
330 IF INKEY$="N" THEN 280
340 GOTO 320
```

EXEMPLE

***** MISE X 0 *****
***** JACK-POT *****
: U

JEUX

PENDU

PRESENTATION

Reproduisant avec une vérité saisissante (!) une véritable potence avec corde et client potentiel (!!), ce programme vous transportera (!!!) quelques années en arrière sur les bancs de la communale.

MODE D'EMPLOI

- 1) Choisir le niveau dont dépendra (!) la longueur du mot à trouver (INKEY\$), soit :

Niveau	Nombre de lettres
1	4
2	5
3	6
4	7
5	8
6	9
7	10
8	11

- 2) Le programme sélectionne un mot dans sa bibliothèque et l'affiche sous forme d'un nombre de coeurs égal au nombre de lettres.
- 3) Indiquer une à une les lettres supposées (INKEY\$). Si la lettre fait partie du mot, elle s'y mettra automatiquement, sinon une nouvelle pièce viendra s'ajouter à la potence.

- 4) Si le mot est trouvé avant que la pendaison ne s'achève, c'est gagné ! S'affiche alors le nombre d'essais ayant mené à la victoire.
- 5) Si le joueur est pendu, le programme affiche le mot juste et propose de rejouer.

N.B. Les mots de la bibliothèque de base peuvent être remplacés par de nouveaux de longueur identique (Lignes 50 à 110).

FONCTIONS

CHR\$, CLS, DATA, DIM, DRAW, ELSE, INKEY\$, LEN, LOCATE, MID\$, READ, RESTORE, RND, VAL.

PROGRAMME

```

5 REM *** PENDU ***
10 CLEAR :CLS :BEEP :PRINT "NIVEAU :
DE 1 A 8 ?":DIM A$(79)*11,G$(1),R$(1)
20 A$=INKEY$:B=VAL(A$):IF B>0 THEN IF
B<9 THEN 40
30 GOTO 20
35 REM *MOTS*
40 CLS :PRINT "UN INSTANT....S.U.P"
50 DATAAIDE,DENT,ROLE,CHAT,MURE,TOLE,
NOIR,VERT,GARE,OCRE
55 DATAECRAN,FROID,GAZON,CLOWN,DOIGT,
TRONC,MYTHE,LOUVE,ECLAT,FRUIT
60 DATAACYPRES,DECLIC,NATURE,PATRIE,SI
MPLE,RAISON,BOUCLE,DETAIL,FORMAT,NOMBRE
70 DATACAPTURE,GAZODUC,ETRANGE,OUTRAG
E,FANTOME,LUMIERE,ANALYSE,REALITE,AMORPH
E
80 DATAHYPNOSE,OBSTACLE,PSYCHOSE,HYST
ERIE,SOUVENIR,PRATIQUE,DISCOURS,TRACTEUR
90 DATASYMPTOME,TRILOGIE,SYNTHESE,PSY
CHISME,SCIATIQUE,CONSCIENT,SOMATIQUE

```

```

95 DATABOURGEOIS,IGNORANCE,MECANIQUE,
COLERIQUE,APATHIQUE,SUBSTANCE
100 DATAPHONETIQUE,OCCUPATION,CENTRIFU
GE,PRIMORDIAL,INCOHERENT,HELICOIDAL
105 DATAEXCAVATION,VILLAGEOIS,ORPHELIN
AT,ANOREXIQUE,PSYCHOLOGIE,AFFECTIVITE
110 DATARECLAMATION,FLEGMATIQUE,PREPAR
ATION,CULPABILITE,CONTINENTAL
115 DATAENIGMATIQUE,TRAVAILLEUR,HYDROM
ETRIE
117 REM *CHARGEMENT*
120 RESTORE 50:FOR X=0 TO 79:READ A$(X
):NEXT X:R$(1)=A$(((B-1)*10+INT(RND*10))
125 S=LEN(R$(1)):FOR K=1 TO S:G$(1)=G$(
1)+CHR$(233):NEXT K:CLS
130 GOSUB 200:BEEP 1:U=U+1:LOCATE 8,1:
PRINT G$(1):IF G$(1)=R$(1) THEN 350
140 E$=INKEY$:IF E$="" THEN 140
150 FOR L=1 TO S:IF MID$(R$(1),L,1)<>E
$ THEN NEXT L:U=U+1:GOTO 130
160 IF L=1 THEN G$(1)=E$+MID$(G$(1),2,
S-1):GOTO 190
170 IF L=S THEN G$(1)=MID$(G$(1),1,S-I
)+E$:GOTO 190
180 G$(1)=MID$(G$(1),1,L-1)+E$+MID$(G$(
1),L+1,S-L)
190 FOR D=L+1 TO S:IF MID$(R$(1),D,1)<
>E$ THEN NEXT D:GOTO 130 ELSE L=D:GOTO 1
60
200 IF U=0 THEN RETURN ELSE GOTO U*10
+200
205 REM *AFFICHAGE DU PENDU*
210 DRAW(0,31)-(40,31):RETURN
220 DRAW(20,31)-(20,0):RETURN
230 DRAW(20,0)-(44,0):RETURN
240 DRAW(44,0)-(44,9):RETURN


```

```

250 LOCATE 5,1:PRINT CHR$(237):GOTO 24
0
260 DRAW(44,15)-(44,22):RETURN
270 DRAW(44,18)-(41,15):DRAW(44,18)-(4
7,15):RETURN
280 DRAW(44,22)-(41,25):DRAW(44,22)-(4
7,25):FOR D=1 TO 300:NEXT D
300 CLS :FOR B=1 TO 10:BEEP 1:BEEP :NE
XT B:PRINT "***** PENDU !! *****C ETAIT
";R$(1)
310 LOCATE 0,2:PRINT "REJOUEZ-VOUS: O/
N ?"
320 IF INKEY$="O" THEN 10
330 IF INKEY$="N" THEN CLS :PRINT "AU
REVOIR,A BIENTOT!":FOR G=1 TO 300:NEXT G
:END
340 GOTO 320
350 FOR N=1 TO 5:BEEP 1:NEXT N:CLS :PR
INT "***** GAGNE !! *****NOMBRE D ESSAIS
=";U
360 GOTO 310

```

EXEMPLE



JEUX

CARNIVAL

PRESENTATION

Apprentissage ludique des codes ASCII, CARNIVAL transforme l'écran du PB 700 en un véritable stand de tir...

MODE D'EMPLOI

- 1) Le jeu consiste à tirer sur des cibles mouvantes (1^{re} ligne) à l'aide d'un pistolet (3^e ligne). Ce dernier se dirige grâce aux touches « 1 » et « 3 ». Le tir s'obtient en pressant la touche « spc » (INKEY\$).
- 2) Les points calculés à partir du code ASCII du caractère touché — 122. Ainsi, a-t-on intérêt à viser ceux dont la valeur est supérieure à 122.
- 3) Chaque tir manqué enlève 10 points. La partie se fait sur 20 passages suivis du score final. Pour rejouer, presser « * » (INKEY\$).

FONCTIONS

ASC, CHR\$, INKEY\$, INT, LOCATE, MID\$, RND, STEP, USING.

PROGRAMME

```

1 REM "*** CARNIVAL ***"
5 CLS :CLEAR :L$=" "+CHR$(144)+" "
:FOR X=1 TO 20

```

```

10 CLS :A$=CHR$(INT(RND*222+33)):Z=IN
T(RND*2):GOSUB 100+Z*10
20 FOR I=C TO D STEP E:LOCATE 0,3:PRI
NT "SCORE:";USING"####";U;
30 LOCATE I,0:PRINT B$:LOCATE J,2:PRI
NT L$:G$=INKEY$
40 IF G$="" THEN NEXT I:NEXT X:GOTO 2
00
50 H=ASC(G$):G$="";IF H>>32 THEN IF H
<>49 THEN IF H>>51 THEN NEXT I:NEXT X:GO
TO 200
60 GOTO H*5
100 B$=" "+A$,C=INT(RND*8):D=18:E=1:T=
1:RETURN
110 B$=A$+" ";C=INT(RND*19):D=0:E=-1:T
=2:RETURN
160 IF J=I-T THEN BEEP 1:U=U-122+ASC(A
$):NEXT X:GOTO 200 ELSE BEEP :U=U-10:GOT
D 40
200 FOR I=1 TO 10:BEEP :BEEP 1:NEXT I
210 CLS :PRINT "***** TERMINE! *****VO
RE SCORE :";U;" PRESSER (*)";
220 IF INKEY$<>"*" THEN 220
230 GOTO 5
245 IF J>=4 THEN J=J-2:GOTO 40 ELSE 40
255 IF J<=13 THEN J=J+2:GOTO 40 ELSE 4
0

```

EXEMPLE

SCORE: 35

JEUX

SUPER MIND

PRESENTATION

Jeu de logique et de déduction, SUPERMIND fera bouillir les méninges !

MODE D'EMPLOI

- 1) Choisir le niveau de jeu (INKEY\$), soit :
CHIFFRES = NIVEAU + 2
(Exemple niveau 5 = 7 chiffres à découvrir).
- 2) Le jeu consiste à découvrir une combinaison cachée en proposant des combinaisons d'un même nombre de termes et en utilisant les réponses du programme. Le programme donne en premier lieu les chiffres bien placés et en second le niveau des chiffres mal placés. Le niveau est obtenu par le cumul des valeurs de comparaison de chaque terme mal placé, soit : +1 si le chiffre est plus grand ou -1 s'il est moins grand.
- 3) Ainsi, par déductions successives, on arrive à découvrir la combinaison cachée.
- 4) Presser « * » pour donner le coup suivant et « O » pour rejouer.

FONCTIONS

CLS, DIM, ELSE, FRAC, INKEY\$, INT, LOCATE, RND, SGN, VAL.

```

5 REM *** SUPER MIND ***
7 CLS :CLEAR :DIM K(20),D$(1)
10 BEEP 1:PRINT "***** OPTION *****"
FORCE : DE 1 A ?"
20 U$=INKEY$:F=VAL(U$):IF F>0 THEN IF
F<8 THEN 50
30 GOTO 20
40 REM *NOMBRE TIRE AU HASARD*
50 F=F+2:A=INT(RND*10^F):C=A:GOSUB 80
55 REM *PROPOSITION*
60 CLS :BEEP 1:PRINT F;" TERMES "::IN
PUT D$(1):IF LEN(D$(1))<>F THEN 60
65 REM *RECHERCHES*
70 B=VAL(D$(1)):I=I+1:C=B:H=F:E=0:D=0
:GOSUB 80:GOTO 120
80 IF A=B THEN 150
90 FOR J=1 TO F
100 K(J+H)=(FRAC(C/10))*10:C=INT(C/10)
:IF K(J)=K(J+H) THEN D=D+1
110 NEXT J:RETURN
120 FOR J=1 TO F:E=E+SGN(K(H+J)-K(J)):
NEXT J
125 REM *INDICATIONS*
130 CLS :BEEP 1:BEEP :PRINT "*****"
*****";E;" ***";D;" ***";I;" ***
"
135 LOCATE 0,2:PRINT " *****"
***"
140 IF INKEY$<>"*" THEN 140 ELSE 60
150 FOR Z=1 TO 3:BEEP :BEEP 1:BEEP :NE
XT Z:CLS
160 PRINT "VOUS AVEZ TROUVE EN",I;" ES
SAIS !!!"
170 BEEP 1:PRINT "REJOUEZ-VOUS: O/N ?"
180 IF INKEY$="O" THEN 7

```

```

190 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR N=1 TO 300:NEXT N:END
200 GOTO 180

```

EXEMPLE

```

***** OPTION *****
FORCE : DE 1 A ?"
1
3 TERMES ?
123
-3 ** 0 ** 1
*
3 TERMES ?
456
-1 ** 0 ** 2
*
3 TERMES ?
789
1 ** 0 ** 3
*
3 TERMES ?
253
-1 ** 0 ** 4
*
3 TERMES ?
267
-1 ** 0 ** 5
*
3 TERMES ?
943
-1 ** 2 ** 6
*
3 TERMES ?
946
-1 ** 2 ** 7

```

*
3 TERMES ?
948
VOUS AVEZ TROUVE EN 8 ESSAIS
REJOUEZ-VOUS: O/N ?
N
AU REVOIR,A BIENTOT!

JEUX

ALLUMETTES

PRESENTATION

Jeu de logique aussi appelé JEU DE NIM, le jeu des allumettes consiste à retirer un certain nombre d'allumettes d'un tas défini au départ et ceci alternativement avec le programme. Celui qui hérite de la dernière a perdu.

MODE D'EMPLOI

- 1) Indiquer le nombre d'allumettes du tas de départ (INPUT).
- 2) Entrer le nombre maximum d'allumettes pouvant être pris à chaque coup (INPUT).
- 3) Préciser si le joueur humain commence, par « O » (INKEY\$).
- 4) Chaque protagoniste retire alternativement une quantité d'allumettes inférieure au montant fixé en 2 (INPUT). Le premier joueur contraint à prendre la dernière allumette a perdu.
- 5) Pour rejouer, presser « O » (INKEYS).

FONCTIONS

ABS, CLS, ELSE, INKEY\$, INT.

PROGRAMME

```
5 REM *** LES MICRO ALLUMETTES ***
7 CLS :CLEAR
10 BEEP 1:INPUT "NOMBRE D ALLUMETTES"
";A:IF A<1 THEN 10
20 BEEP 1:INPUT "PRISE MAXIMALE ";B:I
F B>0 THEN IF B<A THEN 30
25 GOTO 20
30 CLS :BEEP 1:PRINT "COMMENCEZ-VOUS:
O/N ?"
40 C=0:G$=INKEY$: IF G$="N" THEN C=1:G
OTO 60
50 IF G$<>"0" THEN 40
60 BEEP 1:CLS :IF C=1 THEN 90
65 REM *VOTRE COUP*
70 Z=1:BEEP :INPUT "VOTRE COUP ";D:D=
ABSINTD:IF D>0 THEN IF D<=B THEN 80
75 GOTO 20
80 IF D>A THEN 70 ELSE C=1:GOTO 120
85 REM *COUP DE LA MACHINE*
90 E=B+C
95 F=(A-E)/(B+C):IF F=INTF THEN 110
100 E=E-C:GOTO 95
110 D=ABS(E-C):IF D=0 THEN D=C
115 C=0
120 BEEP 1:CLS :IF Z=1 THEN 230 ELSE P
RINT "J EN PRENDS";D
130 Z=0:A=A-D:BEEP 1:PRINT "IL EN REST
E";A
140 FOR X=1 TO 200:NEXT X:IF A=0 THEN
160
150 GOTO 60
160 CLS :FOR H=1 TO 3:BEEP :BEEP 1:BEE
P :NEXT H
```

```
170 IF C>0 THEN PRINT " J AI GAGNE !!
!":GOTO 190
180 PRINT "VOUS AVEZ GAGNE !!!"
190 BEEP :PRINT "REJOUEZ-VOUS: O/N ?"
200 IF INKEY$="O" THEN 7
210 IF INKEY$="N" THEN CLS :PRINT "AU
REVOIR,A BIENTOT!":FOR J=1 TO 300:NEXT J
:END
220 GOTO 200
230 PRINT "VOUS EN PRENEZ";D:GOTO 130
```

EXEMPLE

NOMBRE D ALLUMETTES?
25
PRISE MAXIMALE ?
3
COMMENCEZ-VOUS:O/N ?
0
VOTRE COUP ?
3
VOUS EN PRENEZ 3
IL EN RESTE 22
J EN PRENDS 1
IL EN RESTE 21
VOTRE COUP ?
3
VOUS EN PRENEZ 3
IL EN RESTE 18
J EN PRENDS 1
IL EN RESTE 17
VOTRE COUP ?
3

VOUS EN PRENEZ 3

IL EN RESTE 14

J EN PRENDS 1

IL EN RESTE 13

VOOTRE COUP ?

3

VOUS EN PRENEZ 3

IL EN RESTE 10

J EN PRENDS 1

IL EN RESTE 9

VOOTRE COUP ?

2

VOUS EN PRENEZ 2

IL EN RESTE 7

J EN PRENDS 5

VOOTRE COUP ?

1

UU.

VOUS EN PRENEZ 1

IL EN RESTE 4

J EN PRENDS 3

IL EN RESTE 1

VOOTRE COUP ?

1

VOUS EN PRENEZ 1

IL EN RESTE 0

J AI GAGNE !!!!

REJOUEZ-VOUS: O/N ?

N

AU REVOIR,A BIENTOT!

JEUX

LOGIC 100

PRESENTATION

Jeu de déduction, LOGIC 100 consiste à découvrir un nombre secret en utilisant les indications du programme.

MODE D'EMPLOI

- 1) Choisir l'option de jeu portant sur le nombre d'essais autorisés (INKEY\$).
- 2) Affichage des restes de la division du nombre à découvrir par les chiffres 2, 3, 5 et 7.
- 3) Presser « * » pour proposer un nombre (INKEY\$).
- 4) Entrer la proposition (INPUT). Si celle-ci correspond au nombre mystérieux, le programme affiche le nombre d'essais utilisés.
- 5) Pour rejouer, presser « O » par INKEY.

FONCTIONS

CHR\$, CLS, DIM, ELSE, FRAC, INKEY\$, INT, LOCATE, RND, VAL.

```

5 REM *** LOGIC 100 ***
7 CLS :CLEAR :DIM A(3)
10 BEEP 1:PRINT "***** OPTION *****"
DE 1 A 5 ESSAIS ?
20 G$=INKEY$:A=VAL(G$):IF A>0 THEN IF
A<6 THEN 30
25 GOTO 20
30 CLS :BEEP :N=INT(RND*100+1):A(0)=0
:IF FRAC(N/2)<>0 THEN A(0)=1
35 REM *AFFICHAGE DES RESTES*
40 FOR B=1 TO 3:A(B)=N-(2*B+1)*INT(N/
(2*B+1)):NEXT B:FOR Z=1 TO A
50 CLS :BEEP 1:FOR K=0 TO 3:LOCATE 0,
K
60 PRINT "RESTE PAR";:IF K=0 THEN PRI
NT 2: ELSE PRINT 2*K+1;
70 PRINT "=";:IF A(K)=0 THEN 80 ELSE
FOR G=1 TO A(K):PRINT CHR$(237);:NEXT G
80 NEXT K:FOR M=1 TO 3:BEEP 1:BEEP :N
EXT M
90 IF INKEY$<>"*" THEN 90
95 REM *PROPOSITION*
100 CLS :BEEP :INPUT "VOTRE PROPOSITIO
N ";P:IF P=N THEN 120
105 REM *GAGNE OU PERDU*
110 BEEP 1:CLS :PRINT "***** FAUX !!!
*****":FOR W=1 TO 200:NEXT W:NEXT Z
115 LOCATE 0,1:BEEP :PRINT "***** PERD
U !! ***** C ETAIT ";N:GOTO 130
120 FOR D=1 TO 3:BEEP 1:BEEP :BEEP 1:N
EXT D:CLS :PRINT "***** BRAVO !! *****"
125 LOCATE 0,1:PRINT "TROUVE EN";Z:IF
Z=1 THEN PRINT " ESSAI" ELSE PRINT " ES
SAIS"
130 IF INKEY$<>"*" THEN 130
140 CLS :BEEP :PRINT "REJOUEZ-VOUS: O/
N ?"

```

```

150 IF INKEY$="O" THEN 7
160 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR U=1 TO 300:NEXT U:END
170 GOTO 150

```

EXEMPLE

```

***** OPTION *****
DE 1 A 5 ESSAIS ?
2
RESTE PAR 2=
RESTE PAR 3=o
RESTE PAR 5=oo
RESTE PAR 7=ooooo
*
VOTRE PROPOSITION ?
48
***** FAUX !!! *****
RESTE PAR 2=
RESTE PAR 3=o
RESTE PAR 5=oo
RESTE PAR 7=ooooo
*
VOTRE PROPOSITION ?
82
***** BRAVO !! *****
TROUVE EN 2 ESSAIS
*
REJOUEZ-VOUS: O/N ?
N
AU REVOIR,A BIENTOT!

```

GRAPHO MEMOIRE

PRESENTATION

Jeu graphique, GRAPHO MEMOIRE est un exercice de mémoire visuelle qui plaira tout particulièrement aux jeunes utilisateurs.

MODE D'EMPLOI

- 1) Affichage du tableau (10×3) représentant 30 cartes retournées et identiques deux à deux.
- 2) Entrer les coordonnées de la case à retourner (INKEY\$), soit :
 - Y coordonnée horizontale
 - X coordonnée verticale
- 3) Chaque carte retournée doit être mémorisée et l'on doit essayer d'indiquer à la suite les coordonnées de deux cartes représentant le même graphique. Il faut donc regrouper en un minimum de coups un maximum de cartes.
- 4) Le jeu s'arrête dès que toutes les cartes du tableau ont été retournées.
- 5) Affichage du score final.

FONCTIONS

CHR\$, CLS, DATA, DIM, DRAW INKEY\$, INT, LOCATE, READ, RESTORE, RND, STEP, VAL.

PROGRAMME

```

5 REM *** GRAPHO MEMOIRE ***
10 CLEAR :CLS :DIM B(14),A(2,9):RESTO
RE 500:FOR I=0 TO 14:READ B(I):NEXT I
13 REM GRAPHISMES
15 BEEP 1:LOCATE 2,3:PRINT "012345678
9";
20 FOR A=2 TO 0 STEP -1:LOCATE 0,A:PR
INT 3-A;
30 FOR B=2 TO 11:LOCATE B,A:PRINT CHR
$(142);
35 NEXT B:NEXT A:GOSUB 400:LOCATE 12,
1:BEEP :PRINT "Attendez":LOCATE 12,2
37 REM *CHARGEMENT DU TABLEAU*
40 PRINT "S.v.p. . .":FOR Y=1 TO 15:U=U+
1:FOR T=1 TO 2
41 U=INT(RND*3):W=INT(RND*10):IF A(U,
W)<>0 THEN 41
42 A(U,W)=B(U-1):NEXT T:NEXT Y:BEEP 1
:LOCATE 12,1:PRINT "           ":LOCATE 12,
2
43 PRINT "           ":FOR K=1 TO 10^4:C=
0:LOCATE 13,0:PRINT "           ":GOSUB 200
45 LOCATE X+2,Y:Q=X+2:P=Y:PRINT CHR$(142):GOSUB 400:GOSUB 200
47 IF A(Y,X)=A(P,Q-2) THEN 50
48 LOCATE X+2,Y:PRINT CHR$(142):GOSUB
400:NEXT K
50 LOCATE 13,0:PRINT "-> ";CHR$(A(Y,X
)):C=1:H=X+2:M=Y:GOSUB 200
60 IF X=Q-2 THEN IF Y=P THEN 80
65 REM *PERDU*
70 FOR G=1 TO 3:BEEP :BEEP 1:BEEP :NE
XT G:LOCATE Q,P:PRINT CHR$(142):LOCATE H
,M

```

```

75 PRINT CHR$(142):GOSUB 400:NEXT K
77 REM *GAGNE*
80 FOR G=1 TO 3:BEEP 1:BEEP :BEEP 1:N
EXT G:LOCATE Q,P:PRINT CHR$(A(Y,X))
90 A(M,H-2)=0:A(P,Q-2)=0:FOR S=0 TO 2
:FOR L=0 TO 9:IF A(S,L)<>0 THEN NEXT K
100 NEXT L:NEXT S:FOR G=1 TO 4:BEEP :B
EEP 1:BEEP 1:NEXT G
105 REM *PARTIE TERMINEE*
110 CLS :PRINT "*VOUS AVEZ FINI !!!*NO
MBRE D ESSAIS=";K:LOCATE 0,2
120 BEEP 1:PRINT "REJOUEZ-VOUS; O/N ?"
130 IF INKEY$="O" THEN 10
140 IF INKEY$="N" THEN CLS :PRINT "AU
REVOIR,A BIENTOT!":FOR D=1 TO 300:NEXT D
:END
150 GOTO 130
200 LOCATE 13,1:BEEP :PRINT "*CASE*":L
OCATE 15,2:BEEP 1:PRINT "X ?"
210 N$=INKEY$:X=ASC(N$):IF X>47 THEN I
F X<58 THEN 220
215 GOTO 210
220 X=X-48:BEEP 1:LOCATE 15,2:PRINT "Y
?"
225 N$=INKEY$:Y=VAL(N$):IF Y>0 THEN IF
Y<4 THEN 235
230 GOTO 225
235 Y=3-Y:IF A(Y,X)=0 THEN 200
240 IF C=1 THEN RETURN
250 LOCATE X+2,Y:BEEP :PRINT CHR$(A(Y,
X))
260 FOR I=1 TO 500:NEXT I:RETURN
400 FOR B=7 TO 23 STEP 8:DRAWC(16,B)-((
95,B):NEXT B:RETURN
500 DATA 38,42,43,64,97,111,122,161,17
4,181,185,186,188,219,236

```

EXEMPLE



CASE

JEUX

DIVINATION

PRESENTATION

Devenez devin grâce à ce programme qui mettra à rude épreuve vos facultés PSY !

MODE D'EMPLOI

- 1) Affichage d'une carte suivie de 5 points d'interrogation.
- 2) Presser « + » si l'on suppose que la carte cachée est plus grande ou « - » si plus petite (INKEY\$).
- 3) Chaque bonne réponse donne 10 points, la première erreur arrête la partie.
- 4) Pour rejouer, presser « O » (INKEY\$).

FONCTIONS

CHR\$, CLS, DIM, INKEY\$, LOCATE, MID\$, RND.

PROGRAMME

```
5 REM *** DIVINATION ***
7 CLS :CLEAR :DIM A$(1)
10 BEEP 1:PRINT "***** TOUCHES: *****
(+) POUR PLUS GRAND (-) POUR PLUS PETIT"
```

```
20 FOR K=1 TO 350:NEXT K:CLS :FOR J=1
TO 3:BEEP :BEEP 1:BEEP :NEXT J
25 REM *TIRAGES DES CARTES*
30 A$(1)="234567890UDRA":U=INT(RND*13
+1):C$=CHR$(INT(RND*4+232))
35 W=INT(RND*13+1):D$=CHR$(INT(RND*4+
232)):IF U=W THEN 35
40 REM *AFFICHAGES*
50 LOCATE 9,0:BEEP :PRINT "SCORE=";S
60 LOCATE 0,0:PRINT C$;" ";C$:LOCATE
0,2:PRINT C$;" ";C$
70 LOCATE 1,1:PRINT MID$(A$(1),U,1):L
OCATE 5,0:PRINT "? ?":LOCATE 5,2
75 REM *PROPOSITION*
80 PRINT "? ?":LOCATE 6,1:PRINT "?":F
OR K=1 TO 3:BEEP 1:BEEP :BEEP 1:NEXT K
85 LOCATE 9,1:PRINT "(+) OU (-)"
90 T=0:IF INKEY$="+" THEN T=1:GOTO 11
0
100 IF INKEY$<>"-" THEN 90
110 GOSUB 170:IF W>U THEN IF T=1 THEN
140
120 IF W<U THEN IF T=0 THEN 140
125 REM *PERDU !!
130 FOR K=1 TO 3:BEEP :BEEP 1:BEEP :NE
XT K:LOCATE 9,1:PRINT "*PERDU !!":GOTO
200
140 FOR K=1 TO 3:BEEP 1:BEEP 1:BEEP :N
EXT K
145 REM *GAGNE !!
150 LOCATE 9,1:PRINT "*GAGNE !!":S=S+
10:LOCATE 9,1:FOR J=1 TO 200:NEXT J
160 LOCATE 9,1:PRINT " "":U=W:
C$=D$:GOTO 35
170 LOCATE 5,0:PRINT D$;" ";D$:LOCATE
5,2:PRINT D$;" ";D$:LOCATE 6,1
```

```
180 PRINT MID$(A$(1),W,1):RETURN  
200 FOR F=1 TO 200:NEXT F:CLS :BEEP 1:  
PRINT "REJOUEZ-VOUS: O/N ?"  
210 IF INKEY$="O" THEN 7  
220 IF INKEY$="N" THEN PRINT "AU REVOI  
R, A BIENTOT!":FOR D=1 TO 300:NEXT D:END  
230 GOTO 210
```

EXEMPLE

```
♣ ♣ ? ? SCORE = 0  
4 ? (+) OU (-)  
♣ ♣ ? ?
```

JEUX

CAVERNS OF THORA

PRESENTATION

Jeu graphique par excellence, THORA est une immense caverne d'où il faut sortir sans y laisser sa peau.

MODE D'EMPLOI

- 1) Choisir la taille de la caverne (de 1 à 20 par INKEY\$).
- 2) Affichage de la caverne.
- 3) Le but du jeu est de se frayer un passage dans un dédale de points et ceci avant d'être attrapé par Thoracien and Co. Pour ce faire, on se déplace à l'aide des touches 2, 4, 6 et 8 et l'on tire grâce à la touche « S ».
- 4) La fin de la partie affiche le score final ainsi que toutes les caractéristiques de la partie (temps passé, missiles restants, etc.).
- 5) Pour rejouer, presser « O » (INKEY\$).

FONCTIONS

CHR\$, CLS, DRAW, INKEY\$, INT, LOCATE, POINT, RND.

PROGRAMME

```
5 REM *** CAVERNS OF THORA ***  
10 CLS :CLEAR :BEEP :INPUT "TAILLE DE"
```

```

LA CAVERNE";T:IF T>0 THEN IF TK21 THEN
20
15 GOTO 10
17 REM *FORMATION DU LABYRINTHE*
20 BEEP 1:CLS :FOR I=0 TO T-1:LOCATE
I,0:PRINT CHR$(135):LOCATE I,1:PRINT CHR
$(135)
30 NEXT I:X=0:Y=7:FOR J=0 TO T*8-1:FO
R K=0 TO 15:IF RND>.4 THEN DRAWC(J,K)
40 NEXT K:NEXT J:BEEP 1:LOCATE 0,2:PR
INT "***** PRET? *****":M=INT(RND*T*4+
5)
50 IF INKEY$="" THEN 50
55 REM *AFFICHAGES*
60 LOCATE 0,2:PRINT "*** MISSILES =";
M;" ":DRAW(X,Y)
65 REM *ACTIONS*
70 IF INKEY$="6" THEN IF POINT(X+1,Y)
=0 THEN X=X+1:IF X>=T*8-1 THEN 130
80 IF X>0 THEN IF INKEY$="4" THEN IF
POINT(X-1,Y)=0 THEN X=X-1:IF X<0 THEN X=
0
90 IF Y>0 THEN IF INKEY$="8" THEN IF
POINT(X,Y-1)=0 THEN Y=Y-1:IF Y<0 THEN Y=
0
100 IF INKEY$="8" THEN IF POINT(X,Y-1)
=0 THEN Y=Y-1:IF Y<0 THEN Y=0
110 IF INKEY$="5" THEN IF POINT(X+1,Y)
=1 THEN IF M>0 THEN X=X+1:M=M-1:BEEP
120 D=0+1:GOTO 60
130 FOR G=1 TO 3:BEEP 1:NEXT G:CLS
140 PRINT "SCORE =";INT((1000*T+M*20)/
D); " POINTS"
150 LOCATE 0,1:PRINT "UN AUTRE VOYAGE:
O/N?"
160 IF INKEY$="0" THEN 10

```

```

170 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR D=1 TO 300:NEXT D:END
180 GOTO 160

```

EXEMPLE



GRAPHISME

CERCLES COLORES

PRESENTATION

Exemple d'utilisation des fonctions trigonométriques pour l'impression sur FA-10, ce programme trace des soleils colorés de plusieurs tailles.

MODE D'EMPLOI

- 1) Sélectionner la taille (INKEY\$), de 1 (petit) à 9 (grand).
- 2) Préciser la première couleur d'impression (INKEY\$) soit :

1 : NOIR
2 : BLEU
3 : VERT
4 : ROUGE

- 3) Impression rayon par rayon du cercle coloré.

FONCTIONS

ANGLE, CHR\$, CLS, COS, INKEY\$, LPRINT, SIN, STEP, VAL.

PROGRAMME

```
5 REM *** CERCLES COLORES ***
10 CLS :CLEAR :LPRINT CHR$(28);CHR$(3
22,"F10":ANGLE 1
20 LPRINT "M0,0","M48,0","048,0"
30 BEEP 1:PRINT "TAILLE : DE 1 A 9 ?"
40 G$=INKEY$:T=VAL(G$):IF T>0 THEN IF
T<10 THEN 50
45 GOTO 40
50 BEEP 1:PRINT "COULEUR : DE 1 A 4 ?
"
60 Q$=INKEY$:Z=VAL(Q$):IF Z>0 THEN IF
Z<5 THEN 70
65 GOTO 60
70 REM *DESSIN*
80 C=1:U=T*5:Z=Z-1
90 FOR I=0 TO 6.28 STEP .1
100 X=COS(I)*U:Y=SIN(I)*U
110 IF I=1.6*C THEN C=C+1:Z=Z+1
120 IF Z>3 THEN Z=0
130 LPRINT "J";Z,"D0,0,";X;";Y:NEXT
I:LPRINT "F6"
140 BEEP 1:CLS :PRINT "UN AUTRE DESSIN
:O/N?"
150 IF INKEY$="O" THEN 10
155 IF INKEY$<>"N" THEN 150
160 PRINT "AU REVOIR,A BIENTOT!":FOR H
=1 TO 300:NEXT H:ANGLE 0:LPRINT "00,0",
F8":END
170 GOTO 150
```


EXEMPLE

TAILLE : DE 1 A 9 ?

3

COULEUR : DE 1 A 4 ?

1



UN AUTRE DESSIN: O/N ?

N

AU REVOIR, A BIENTOT!

GRAPHISME

ELLIPSSES COLOREES

PRESENTATION

Tout comme CERCLES COLORES, le programme ELLIPSES illustre les nombreuses possibilités graphiques du PB 700 et particulièrement l'usage des fonctions trigonométriques.

MODE D'EMPLOI

- 1) Choisir la taille de l'ellipse (INKEY\$), de 1 (petit) à 9 (grand).
- 2) Sélectionner la couleur qui servira au tracé du dessin (INKEY\$), soit :

1 : NOIR
2 : BLEU
3 : VERT
4 : ROUGE

- 3) Indiquer le coefficient d'aplatissement de l'ellipse (angle d'inclinaison), de 1 à 5 (INKEY\$).
- 4) Tracé de l'ellipse.

FONCTIONS

CHR\$, COS, INKEY\$, LPRINT, SIN, VAL.

PROGRAMME

```
5 REM *** ELLIPSES COLOREES ***
10 CLEAR :CLS :LPRINT CHR$(28);CHR$(3
20,"F8":ANGLE 1
20 LPRINT "M0,0","M48,0","048,0"
30 BEEP 1:PRINT "TAILLE : DE 1 A 9 ?"
40 G$=INKEY$:T=VAL(G$):IF T>0 THEN IF
T<10 THEN 50
45 GOTO 40
50 BEEP 1:PRINT "COULEUR : DE 1 A 4 ?"

60 Q$=INKEY$:Z=VAL(Q$):IF Z>0 THEN IF
Z<5 THEN 70
65 GOTO 60
70 Z=Z-1:BEEP 1:PRINT "APLATISSEMENT:
1-5 ?"
72 K$=INKEY$:R=VAL(K$):IF R>0 THEN IF
R<6 THEN 80
75 GOTO 72
80 U=T*5:W=T*R
90 FOR I=0 TO 6.28 STEP .07
100 X=COS(I)*U:Y=SIN(I)*W
130 LPRINT "J";Z,"D0,0";X,"";Y:NEXT
I:LPRINT "F5"
140 BEEP 1:CLS :PRINT "UN AUTRE DESSIN
:O/N?"
150 IF INKEY$="O" THEN 10
155 IF INKEY$<>"N" THEN 150
160 PRINT "AU REVOIR,A BIENTOT!":FOR H
=1 TO 300:NEXT H:ANGLE 0:LPRINT "F5","00
,0":END
170 GOTO 150
```

EXEMPLE

```
TAILLE : DE 1 A 9 ?
3
COULEUR : DE 1 A 4 ?
1
APLATISSEMENT 1-5 ?
2
```



UN AUTRE DESSIN:O/N?
N
AU REVOIR,A BIENTOT!

GRAPHISME TELECRAN

PRESENTATION

Inspiré du célèbre jeu du même nom, ce programme permet la commande directe d'un curseur d'affichage qui pourra être librement dirigé et réalisera ainsi toutes les figures que votre imagination vous dictera.

MODE D'EMPLOI

- 1) Affichage de « VOUS POUVEZ DESSINER » qui signifie que l'on peut dessiner directement sur l'écran en manipulant les touches 1 à 4 et 6 à 9 (INKEY\$).
- 2) On peut, grâce aux touches « * » et « 5 », modifier le tracé du curseur (INKEY\$), soit :
 - pour allumer les points où le curseur passe (son aigu)
 - 5 pour éteindre les points où le curseur passe (son grave)
- 3) Pour arrêter le programme, presser « F » par INKEY\$.

FONCTIONS


CLS, DRAW, ELSE, INKEY\$.

PROGRAMME

```
5 REM ***TELECRAN***  
7 CLS :CLEAR :X=0:Y=0:I=160
```

```
10 FOR G=1 TO 3:BEEP :BEEP 1:NEXT G  
20 PRINT "VOUS POUVEZ DESSINER":FOR K  
=1 TO 200:NEXT K:BEEP :CLS  
25 REM*ACTIONS*  
30 F$=INKEY$:IF F$="5" THEN C=1  
35 IF F$=".+" THEN C=0  
40 IF F$="1" THEN X=X-1:Y=Y+1:GOSUB I  
50 IF F$="2" THEN Y=Y+1:GOSUB I  
60 IF F$="3" THEN Y=Y+1:X=X+1:GOSUB I  
70 IF F$="4" THEN X=X-1:GOSUB I  
80 IF F$="6" THEN X=X+1:GOSUB I  
90 IF F$="7" THEN X=X-1:Y=Y-1:GOSUB I  
100 IF F$="8" THEN Y=Y-1:GOSUB I  
110 IF F$="9" THEN Y=Y-1:X=X+1:GOSUB I  
115 REM *DESSIN*  
120 IF C=1 THEN BEEP :DRAWC(A,B):DRAWC  
X,Y) ELSE DRAW(X,Y):BEEP 1  
130 IF F$="F" THEN CLS :END  
150 A=X:B=Y:GOTO 30  
155 REM *VERIFICATIONS*  
160 IF X<0 THEN X=0  
170 IF Y<0 THEN Y=0  
180 IF X>159 THEN X=159  
190 IF Y>31 THEN Y=31  
200 RETURN
```

EXEMPLE



GRAPHISMES DIVERS

PRESENTATION

Illustration des possibilités de l'écran graphique, ce programme trace, en plusieurs tailles, des formes simples telles que les cercles ou les triangles. On pourra très aisément l'intégrer comme sous-programme d'une représentation plus évoluée.

MODE D'EMPLOI

1) Sélectionner la forme à dessiner (INKEY\$), soit :

- 1 : Rectangles
- 2 : Ronds
- 3 : Carrés
- 4 : Triangles

2) Préciser la taille du graphisme choisi en 1 (INKEY\$).

3) Réalisation point par point du dessin, presser « * » pour continuer et « O » pour une autre représentation (INKEY\$).

FONCTIONS

CLS, COS, DRAW, INKEY\$, SIN, STEP, TAB, VAL.

PROGRAMME

```

5 REM *** GRAPHISMES DIVERS ***
7 CLEAR :CLS
10 PRINT "1:RECTANGLE    2:RONDE3:CARRE
4:TRIANGLE":BEEP 1
20 A$=INKEY$:B=VAL(A$):BEEP :IF B>0 T
HEN IF B<5 THEN 40
30 GOTO 20
40 CLS :PRINT "TAILLE : DE 1 A 5 ?"
50 A$=INKEY$:C=VAL(A$):IF C>0 THEN IF
C<6 THEN CLS :GOTO B*100
60 GOTO 50
100 L=32*C:H=6*C:X=(160-L)/2:Y=(32-H)/
2:IF L=160 THEN L=159
120 DRAW(X,Y)-(X+L,Y)-(X+L,Y+H)-(X,Y+H
)-(X,Y):GOTO 500
200 X=3*C:Y=3*C
210 FOR A=0 TO 360 STEP 12:DRAW(SINA**X
+80,COSA*Y+15):NEXT A:GOTO 500
300 L=6*C:H=L:X=(160-L)/2:Y=(32-H)/2:G
OTO 120
400 L=32*C:X=(160-L)/2:H=(32-6*C)
410 DRAW(X,31)-(X+L,31)-(X+(L/2),H)-(X
,31):BEEP
500 IF INKEY$<>"*" THEN 500
600 CLS :PRINT TAB(20);"UN AUTRE DESSI
N:O/N?"
610 IF INKEY$=="O" THEN 7
620 IF INKEY$=="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR G=1 TO 300:NEXT G:END
630 GOTO 610

```

EXEMPLE



GRAPHISME

TRIANGLES DIVERS

PRESENTATION

Si la manipulation du FA-10 n'est pas toujours aisée, elle en sera sûrement facilitée par l'exemple graphique suivant qui trace de simples triangles de tailles et de dessins variables.

MODE D'EMPLOI

- 1) Choisir la taille du triangle à imprimer (INKEY\$), soit :
1 : (petit) à 3 (grand)
2) Préciser la couleur d'impression (INKEY\$), soit :
1 : pour noir
2 : pour bleu
3 : pour vert
4 : pour rouge.
3) Indiquer la forme du trait (INKEY\$), soit :
1 : pour trait plein
2 : pour trait cassé
3 : pour trait à une chaîne
4 : pour trait à deux chaînes.
4) Impression du triangle.

FONCTIONS

CHR\$, CLS, INKEY\$, LPRINT, VAL.

PROGRAMME

```
5 REM *** TRIANGLES DIVERS ***
10 CLEAR :CLS :LPRINT CHR$(28);CHR$(3
2)
20 BEEP 1:PRINT "TAILLE : DE 1 A 9 ?"
30 G$=INKEY$:T=VAL(G$):IF T>0 THEN IF
T<10 THEN 40
35 GOTO 30
40 BEEP 1:PRINT "COULEUR : DE 1 A 4 ?
"
50 G$=INKEY$:C=VAL(G$):IF C>0 THEN IF
C<5 THEN 70
60 GOTO 50
70 C=C-1:O=T*2:BEEP 1:PRINT "FORME DU
TRAIT:1-4 ?"
80 G$=INKEY$:L=VAL(G$):IF L>0 THEN IF
L<5 THEN 100
85 GOTO 80
90 REM *TRACE DU TRINGLE*
100 L=L-1:LPRINT "M33,0","033,0":F=-0/
2
120 LPRINT "J";C,"L";L,"D0,0,";F;",";F
;";O;";F;";0,0"
130 LPRINT "M-33,0","0-33,0"
140 CLS :BEEP 1:PRINT "UN AUTRE DESSIN
: O/N?"
150 IF INKEY$="O" THEN 10
160 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR Y=1 TO 300:NEXT Y:END
170 GOTO 150
```

EXEMPLE

```
TAILLE : DE 1 A 9 ?
3
COULEUR : DE 1 A 4 ?
1
FORME DU TRAIT:1-4 ?
4
```



UN AUTRE DESSIN: O/N?
N
AU REVOIR, A BIENTOT!

CARRES DIVERS

PRESENTATION

Utilisable seul ou en sous/programme, CARRES DIVERS se charge de mettre en œuvre un maximum de fonctions d'impression pour la réalisation de carrés de graphisme totalement libre.

MODE D'EMPLOI

1) Sélectionner la taille du carré à dessiner (de 1 à 9, INKEY\$).

2) Choisir la couleur du dessin (INKEY\$), soit :

- 1 : pour noir
- 2 : pour bleu
- 3 : pour vert
- 4 : pour rouge.

3) Préciser le type de hachures souhaité (INKEY\$), soit :

- 1 : pas de hachure
- 2 : hachures horizontales
- 3 : hachures verticales.

4) Entrer l'écartement des hachures (INKEY\$), soit :

1 (peu écartées) à 9 (très écartées).


5) Dessin automatique du carré en fonction des choix faits en 1, 2, 3 et 4.

PROGRAMME

```
5 REM *** CARRÉS DIVERS ***
10 CLEAR :CLS :LPRINT CHR$(28);CHR$(3
20
20 BEEP 1:PRINT "TAILLE : DE 1 A 9 ?"
30 G$=INKEY$:T=VAL(G$):IF T>0 THEN IF
T<10 THEN 40
35 GOTO 30
40 BEEP 1:PRINT "COULEUR : DE 1 A 4 ?
"
50 G$=INKEY$:C=VAL(G$):IF C>0 THEN IF
C<5 THEN 70
60 GOTO 50
70 C=C-1:BEEP 1:PRINT "HACHURES: DE 1
A 3 ?"
80 G$=INKEY$:Z=VAL(G$):IF Z>0 THEN IF
Z<4 THEN 90
85 GOTO 80
90 BEEP 1:Z=Z-1:PRINT "ECARTEMENT:DE
1 A 4 ?"
100 F$=INKEY$:E=VAL(F$):IF E>0 THEN IF
E<5 THEN 110
105 GOTO 100
107 REM *TRACE DU CARRÉ*
110 X=T*5:LPRINT "050,-25"
120 LPRINT "J";C,"A";-X;",";X;",";X;",
";-X,"G";Z;",";X*2;",";-X*2;";E
140 CLS :BEEP 1:PRINT "UN AUTRE DESSIN
:O/N?"
150 IF INKEY$="O" THEN 10
160 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR Y=1 TO 300:NEXT Y:END
170 GOTO 150
```

EXEMPLE

```
TAILLE : DE 1 A 9 ?  
3  
COULEUR : DE 1 A 4 ?  
1  
HACHURES: DE 1 A 3 ?  
2  
ECARTEMENT:DE 1 A 4?  
4
```



UN AUTRE DESSIN:O/N?

N

AU REVOIR,A BIENTOT!

GRAPHISME

INVERSION VIDEO

PRESENTATION

Parfaite illustration de l'utilisation de « POINT » et de « DRAW, DRAWC », l'inversion vidéo transforme l'écran en son homologue négatif.

MODE D'EMPLOI

- 1) Entrer un nombre de signes inférieur à 60 (INPUT).
- 2) Tracé ligne par ligne du négatif.
- 3) Presser « » pour continuer, et « O » pour une nouvelle inversion, sinon « N » (INKEY\$).

FONCTIONS

CLS, DIM, DRAW, INKEY\$, LEN, POINT.

PROGRAMME

```
5 REM *** INVERSION VIDEO ***  
10 CLS :CLEAR :DIM A$(1)*59  
20 INPUT "CARACTERES (MAX=59)":A$(1):  
IF LEN(A$(1))>59 THEN 20  
30 CLS :PRINT A$(1);  
33 REM *INVERSION*  
35 FOR L=0 TO 159:FOR H=0 TO 31
```

```
40 P=POINT(L,H): IF P=1 THEN DRAWC(L,H)
50 IF P=0 THEN DRAW(L,H)
60 NEXT H:NEXT L:BEEP
70 IF INKEY$<>"*" THEN 70
80 CLS :PRINT "AUTRE INVERSION: O/N?"
90 IF INKEY$="O" THEN 10
100 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR J=1 TO 300:NEXT J:END
110 GOTO 90
```

EXEMPLE

URGÉ, DES ESPoir, vieillesse ennemie,
n' ai je donc tant..

MATHEMATIQUES

ARABES-ROMAINS

PRESENTATION

A l'attention des matheux nostalgiques, voici un programme qui transformera tous les nombres exprimés en chiffres arabes en chiffres romains et inversement.

MODE D'EMPLOI

- 1) Entrer un nombre, soit romain, soit arabe, suivi de RETURN.
- 2) Le programme traduit automatiquement dans l'autre système le nombre rentré en 1.
- 3) Affichage de la traduction, presser « * » pour continuer et répondre « O » pour refaire une conversion (INKEY\$).

FONCTIONS

ASC, CLS, DIM, INKEY\$, INT, LEN, MID\$, VAL.

PROGRAMME

```
5 REM *** CONVERSION ARABE <--> ROM
AIN ***
7 CLEAR :DIM N$(1)*30
10 E=0:U=10:CLS :BEEP :INPUT "VOTRE N
OMBRE ";N$(1):FOR I=1 TO LEN(N$(1))
```

```

15 REM *TRAITEMENT DES CHAINES*
20 IF MID$(N$(1),I,1)=" " THEN NEXT I
:GOTO 10
30 N$(1)=MID$(N$(1),I,LEN(N$(1))-I+1)
:FOR J=1 TO LEN(N$(1))
40 A$=MID$(N$(1),J,1):GOSUB 50:Z=U*Z+
R:NEXT J:GOTO 160
50 IF ASC(A$)>47 THEN IF ASC(A$)<58 T
HEN R=VAL(A$):RETURN
60 G=5040:FOR K=1 TO LEN(N$(1)):A$=MI
D$(N$(1),K,1):H=1
70 IF A$="U" THEN H=5
80 IF A$="X" THEN H=U
90 IF A$="L" THEN H=50
100 IF A$="C" THEN H=100
110 IF A$="D" THEN H=500
120 IF A$="M" THEN H=1000
130 IF H=1 THEN IF A$<>"I" THEN 10
140 F=H:IF H>G THEN .F=H-2*G
145 REM *RESULTAT EN ARABE*
150 G=H:E=E+F:NEXT K:BEEP 1:CLS :PRINT
N$(1);" =";E:GOTO 340
160 BEEP 1:CLS :PRINT N$(1);" = ";
170 IF Z>999 THEN PRINT "M";:Z=Z-1000:
GOTO 120
180 FOR L=1 TO 3:GOSUB L*10+300:B=INT(
Z/U^(3-L))
190 GOSUB B*10+210
200 Z=Z-B*U^(3-L):NEXT L:GOTO 340
205 REM *RESULTAT EN ROMAIN*
210 RETURN
220 PRINT N$(1)::RETURN
230 PRINT N$(1);N$(1)::RETURN
240 PRINT N$(1);N$(1);N$(1)::RETURN
250 PRINT N$(1);X$::RETURN
260 PRINT X$::RETURN
270 PRINT X$;N$(1)::RETURN

```

```

280 PRINT X$::GOTO 230
290 PRINT X$::GOTO 240
300 PRINT N$(1);Y$::RETURN
310 N$(1)="C":X$="D":Y$="M":RETURN
320 Y$=N$(1):N$(1)="X":X$="L":RETURN
330 Y$=N$(1):N$(1)="I":X$="U":RETURN
340 IF INKEY$<>"*" THEN 340
350 CLS :PRINT "UN AUTRE CALCUL:O/N?"
360 IF INKEY$="O" THEN 2
370 IF INKEY$="N" THEN PRINT TAB(20);"
AU REVOIR,A BIENTOT!":FOR X=1 TO 300:NEX
T X:END
380 GOTO 360

```

EXEMPLE

3	1	5	4	8	=	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	M	D
X	L	V	I	I	I																									

MATHEMATIQUES

TEMPERATURES

PRESENTATION

Exprimées dans de multiples unités, les températures seront facilement converties par l'intermédiaire de ce programme.

MODE D'EMPLOI

- 1) Indiquer l'unité de départ, et celle d'arrivée, en présentant leur numéro respectif.
- 2) Introduire dans l'unité choisie la température à convertir (INPUT).
- 3) Affichage de la conversion. Presser « O » pour continuer (INKEY\$).

FONCTIONS

CLS, DATA, DIM, INKEY\$, LOCATE, READ.

PROGRAMME

```
2 CLEAR :CLS :DIM A$(5)*20,X(2)
10 DATADEGRES FARENHEIT =,DEGRES CELSIUS =,DEGRES KELVIN =
11 DATADEGRES REAUMUR =,DEGRES RANKIN E =
```

```
15 FOR G=1 TO 5:READ A$(G):NEXT G:M=1
60
20 CLS :FOR I=0 TO 1:PRINT "OPTION":I+1;"*":LOCATE 9,0
30 PRINT "1:FARENHEIT2:CELSIUS / 3:KE
LUIN4:REAUMUR/5:RANKINE"
35 BEEP :D$=INKEY$:U=VAL(D$):IF U>0 THEN IF U<6 THEN 50
40 GOTO 35
50 X(I)=U:CLS :NEXT I:C=X(1)*10+100:G
OTO X(0)*10+50
55 REM *CALCULS ET RESULTATS*
60 PRINT A$(1)::INPUT R:GOTO C
70 PRINT A$(2)::INPUT T:R=T*(9/5)+32:
GOTO C
80 PRINT A$(3)::INPUT T:R=(T-273.15)*
(9/5)+32:GOTO C
90 PRINT A$(4)::INPUT T:R=T*(9/4)+32:
GOTO C
100 PRINT A$(5)::INPUT T:R=T-459.58:GO
TO C
110 PRINT A$(1);R:GOTO M
120 F=(R-32)*(5/9):PRINT A$(2);F:GOTO
M
130 F=(R-32)*(5/9)+273.15:PRINT A$(3);
F:GOTO M
140 F=(R-32)*(4/9):PRINT A$(4);F:GOTO
M
150 F=R+459.58:PRINT A$(5);F
160 BEEP :IF INKEY$<>"*" THEN 160
170 CLS :PRINT "UN AUTRE CALCUL:O/N?"
180 IF INKEY$="O" THEN 20
190 IF INKEY$="N" THEN PRINT TAB(20);"
AU REVOIR,A BIENTOT!":FOR J=1 TO 300:NEX
T J:END
200 GOTO 180
```

EXEMPLE

```
OPTION1*1:FARENHEIT  
2:CELSUIS 3:KELVIN  
4:REAUMUR 5:RANKINE
```

1

```
OPTION2*1:FARENHEIT  
2:CELSIUS 3:KELVIN  
4:REAUMUR 5:RANKINE
```

2

DEGREE FARENHEIT ?

10

DEGREE CELSIUS=-12.2
22

MATHEMATIQUES

VOLUMES

PRESENTATION

Ce programme calcule les volumes des formes suivantes : cubes, parallélépipèdes rectangles, cylindres, sphères et pyramides.

MODE D'EMPLOI

1) Choisir l'option par INKEY\$, soit :

- 1 : pour cube
- 2 : pour parallélépipède rectangle
- 3 : pour cylindre
- 4 : pour sphère
- 5 : pour pyramide.

2) Entrer les données permettant le calcul (INPUT).

3) Affichage du résultat. Presser « O » pour continuer (INKEY\$).

FONCTIONS

CLS, INKEY\$, LOCATE, PI, VAL.

PROGRAMME

```
5 REM *** VOLUMES DIVERS ***
10 CLS :CLEAR :BEEP 1:PRINT "***** 0
PTION *****:(1) (2) (3) (4) (5)"
30 G$=INKEY$:A=VAL(G$):IF A>0 THEN IF
A<6 THEN CLS :BEEP 1:GOTO 20*A+40
40 GOTO 30
55 REM *CUBE*
60 INPUT "COTE DU CUBE ";C
70 U=C^3:GOTO 160
75 REM *PLP RECTANGLE*
80 INPUT "LONGUEUR DU PLP ";L,"SA LAR
GEUR ";N,"SA HAUTEUR ";H
90 U=H*L*N:GOTO 160
95 REM *CYLINDRE*
100 INPUT "RAYON DU CYLINDRE ";R,"SA H
AUTEUR ";H
110 U=PI*R^2*H:GOTO 160
115 REM *SPHERE*
120 INPUT "RAYON DE LA SPHERE ";R
130 U=(4/3)*PI*R^3:GOTO 160
135 REM *PYRAMIDE*
140 INPUT "BASE DE LA PYRAMIDE";B,"SA
HAUTEUR ";H
150 U=(1/3)*B*H
155 REM *RESULTAT*
160 CLS :BEEP :PRINT "VOLUME =";U:LOCA
TE 0,1
170 PRINT "UN AUTRE CALCUL:O/N?"
180 IF INKEY$="O" THEN 10
190 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR G=1 TO 300:NEXT G:END
200 GOTO 180
```

EXEMPLE

```
***** OPTION *****
(1) (2) (3) (4) (5)
3
RAYON DU CYLINDRE ?
25
SA HAUTEUR ?
10
VOLUME = 19634.95409
UN AUTRE CALCUL:O/N?
0
***** OPTION *****
(1) (2) (3) (4) (5)
5
BASE DE LA PYRAMIDE?
15
SA HAUTEUR ?
25
VOLUME = 125
UN AUTRE CALCUL:O/N?
N
AU REVOIR,A BIENTOT!
```

MATHÉMATIQUES

POLAIRE-RECTANGLE

PRESENTATION

Ce programme convertit les coordonnées polaires en coordonnées rectangulaires et inversement.

MODE D'EMPLOI

1) Choisir l'option de conversion, (INKEY\$), soit :

- 1 : pour polaire vers rectangulaire
- 2 : pour rectangulaire vers polaire

2) Entrer chaque coordonnée dans le système de départ (INPUT).

3) Affichage de la conversion.

4) Pour continuer, presser « * » (INKEY\$).

FONCTIONS

ATN, CLS, COS, ELSE, INKEY\$, PI, SIN, SQR, VAL.

PROGRAMME

```
5 REM *** COORDONNEES RECTANGULAIRES
<-> POLAIRES ***
7 CLS :CLEAR :BEEP 1
```

```
10 PRINT "***** OPTION *****(1):POL
AIRE --> RECT(2):RECT -->POLAIRE"
20 C$=INKEY$: U=VAL(C$): IF U>0 THEN IF
U<3 THEN GOTO U*30
25 GOTO 20
27 REM *POL->RECT*
30 BEEP 1:CLS :INPUT "ALPHA (EN DEGRE
S) ";A,"BETA (EN DEGRES) ";B
40 D=B*PI/180:X=A*COS(D):Y=A*SIN(D)
50 CLS :BEEP :PRINT "X =";X,"Y =";Y
55 IF INKEY$<>"*" THEN 55 ELSE 140.
57 REM *RECT->POL*
60 CLS :BEEP 1:INPUT "COORDONNEE X ";
X,"COORDONNEE Y ";Y
70 A=SQR(X^2+Y^2):IF A=0 THEN B=0:GOT
O 130
80 IF X<>0 THEN 100
90 IF Y>0 THEN B=90:GOTO 130
95 B=220:GOTO 130
100 D=ATN(Y/X)*180/PI:IF X<0 THEN D=D+
180
110 IF D<0 THEN D=D+360
120 B=D
130 CLS :BEEP :PRINT "ALPHA =";A,"BETA
=";B:GOTO 55
140 BEEP 1:CLS :PRINT "UN AUTRE CALCUL
:O/N?"
150 IF INKEY$="O" THEN 7
160 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR G=1 TO 300:NEXT G:END
170 GOTO 150
```

EXEMPLE

```
***** OPTION *****  
(1):POLAIRE --> RECT  
(2):RECT --> POLAIRE  
1  
ALPHA (EN DEGRES) ?  
25  
BETA (EN DEGRES) ?  
38  
X=19.70026884  
Y=15.39153688  
*  
UN AUTRE CALCUL:O/N?  
0  
***** OPTION *****  
(1):POLAIRE --> RECT  
(2):RECT --> POLAIRE  
2  
COORDONNEE X ?  
36  
COORDONNEE Y ?  
2.5  
ALPHA= 36.08670115  
BETA= 3.972495941  
*  
UN AUTRE CALCUL:O/N?  
N  
AU REVOIR,A BIENTOT!
```

MATHEMATIQUES

NOMBRES PREMIERS

PRESENTATION

Réputés comme étant indivisibles, les nombres premiers seront facilement mis à jour par ce programme qui, de plus, décompose les nombres divisibles en facteurs premiers.

MODE D'EMPLOI

- 1) Choisir l'option par INKEY\$, soit :
 - 1 : pour déterminer si un nombre est premier
 - 2 : pour afficher tous les nombres premiers entre deux bornes.
- 2) Dans les deux cas, préciser les nombres intervenant dans la détermination (INPUT).
- 3) Affichage du résultat, presser « * », pour continuer (INKEY\$).

FONCTIONS

ABS, CLS, ELSE, FRAC, INKEY\$, INT, SQR, STEP, VAL.

PROGRAMME

```
5 REM *** NOMBRES PREMIERS ***  
2 CLS :CLEAR
```

```

10 BEEP 1:PRINT "***** OPTION *****"
1:N EST-IL PREMIER ?2:RECHERCHE LIMITEE"
20 G$=INKEY$:U=VAL(G$):IF U>0 THEN IF
U<3 THEN GOTO U*100-70
25 GOTO 20
27 REM *NOMBRE PREMIER ?*
30 CLS :BEEP 1:INPUT "VOTRE NOMBRE ";
B:CLS :B=ABS(INT(B)):IF B<4 THEN PRINT B
::GOTO 100
40 D=SQR(B):FOR C=2 TO D
50 IF FRAC(B/C)=0 THEN H=H+1:IF H=1 T
HEN PRINT "DIVISEURS ="
60 IF FRAC(B/C)=0 THEN PRINT C;:B=B/C
:C=C-1:F=1
70 IF B=1 THEN C=D
80 NEXT C:IF B>1 THEN PRINT B;
90 IF F=1 THEN 200
100 BEEP :PRINT " EST PREMIER":GOTO 20
0
120 REM *RECHERCHE LIMITEE*
130 CLS :BEEP 1:INPUT "1ER NOMBRE ";A,
"2EME NOMBRE ";B:A=INT(A):B=INT(B):CLS
135 IF A>0 THEN IF B>A THEN 140 ELSE 1
30
140 IF A<3 THEN X$=" 2":A=A+1:IF A=2 T
HEN A=3:X$="1 2"
150 BEEP :PRINT "ENTRE CES 2 BORNES:",X$::IF FRAC(A/2)=0 THEN A=A+1
160 FOR C=A TO B STEP 2:D=SQR(C):FOR E
=2 TO D
170 IF FRAC(C/E)=0 THEN E=D:F=1
180 NEXT E:IF F=0 THEN PRINT C;:P=1
190 F=0:NEXT C:IF P=0 THEN PRINT "AUCU
N !!"
200 FOR S=1 TO 3:BEEP 1:NEXT S
205 IF INKEY$<>"*" THEN 205

```

```

210 CLS :BEEP :PRINT "UN AUTRE CALCUL:
O/N?"
220 IF INKEY$="O" THEN 2
230 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR G=1 TO 300:NEXT G:END
240 GOTO 220

```

EXEMPLE

```

***** OPTION *****
1:N EST-IL PREMIER?
2:RECHERCHE LIMITEE
1
VOTRE NOMBRE ?
2563
DIVISEURS:
11 233
*
UN AUTRE CALCUL:O/N?
0
***** OPTION *****
1:N EST-IL PREMIER?
2:RECHERCHE LIMITEE
2
1ER NOMBRE ?
125427
2EME NOMBRE ?
125474
ENTRE CES 2 BORNES:
125429 125441
125453 125471
*
UN AUTRE CALCUL:O/N?
N
AU REVOIR,A BIENTOT!

```

MATHEMATIQUES

FRACTIONS

PRESENTATION

Ce programme permet de réduire les fractions réductibles et de transformer des nombres décimaux en fractions d'entiers.

MODE D'EMPLOI

1) Choisir l'option (INKEY\$), soit :

- 1 : pour approximation
- 2 : pour réduction

- 2) Approximation : entrer la valeur décimale que l'on désire transformer en fraction de nombres entiers (INPUT) ainsi que la précision de calcul correspondant au nombre de chiffres après la virgule pris en compte pour l'approximation.
- 3) Réduction : entrer le numérateur A et le dénominateur B (INPUT). Affichage de la fraction réduite (si cela est possible).
- 4) Dans tous les cas, on continue le déroulement en pressant « * » après l'affichage des résultats.

FONCTIONS

ABS, CLS, FRAC, INKEY\$, INT, VAL.

PROGRAMME

```
5 REM *** CALCULS SUR FRACTIONS ***
7 CLS :CLEAR :F$="A/B =": J=10
10 BEEP* 1:PRINT "***** OPTION *****"
(1) APPROXIMATION (2) REDUCTION"
20 G$=INKEY$:H=VAL(G$):IF H=2 THEN CL
S :BEEP 1:GOTO 100
25 IF HK>1 THEN 20
30 CLS :BEEP 1:INPUT "VOTRE NOMBRE ";
X,"VOTRE PRECISION ";Z:Z=J^Z:P=FRACX
40 A=0:B=1:IF P<>0 THEN 60
50 N=X:C=B:GOTO 80
60 D=1/P:C=B*INT(D)+A:A=B:B=C:N=INT(X
*C+.5):P=FRAC(D)
70 IF INT((N/C-X)*Z)/Z<>0 THEN 60
80 BEEP :CLS :PRINT F$;N;" /";C:GOTO
255
100 D=0:P=10^-6:I=1:E=-1
110 INPUT "A/B : NUMERATEUR ";X,"A/B :
DENOMINATEUR ";Y:A=X:B=Y:CLS
120 IF B<>0 THEN 180 ELSE 110
130 H=INT(A):O=(A-H)/J:IF O=0 THEN 160
140 Z=Z*K:E=E+1:IF O<.1 THEN 140
160 I=I+1:T=INT(O*I+.4)-O*I:IF PK=ABST
THEN 160
170 A=INT((Z/(J^E)+K)*I*(J^E)+.4):B=I*
J^E:P=B:H=A:D=1
180 IF BK=A THEN 200
190 C=A:A=B:B=C
200 Q=INT(A/B):R=A-B*Q:IF R=0 THEN 220
210 A=B:B=R:GOTO 200
220 IF D=0 THEN 240
230 BEEP :PRINT F$;H/B;" /";P/B:GOTO 2
```

```

240 IF Y/B=1 THEN BEEP 1:PRINT F$;X/B:
GOTO 255
250 BEEP 1:PRINT F$;X/B;" /";Y/B
255 IF INKEY$<>"*" THEN 255
260 CLS :BEEP 1:PRINT "UN AUTRE CALCUL
:O/N?"
270 IF INKEY$="O" THEN 2
280 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR G=1 TO 300:NEXT G:END
290 GOTO 270

```

EXEMPLE

***** OPTION *****	***** OPTION *****
(1) APPROXIMATION	(1) APPROXIMATION
(2) REDUCTION	(2) REDUCTION
1	2
VOITRE NOMBRE ?	A/B : NUMERATEUR ?
12.5361	152
VOITRE PRECISION ?	A/B : DENOMINATEUR ?
4	28
A/B = 6945 / 554	A/B = 38 / 2
*	*
UN AUTRE CALCUL:O/N?	UN AUTRE CALCUL:O/N?
0	N
	AU REVOIR,A BIENTOT!

MATHEMATIQUES

PGCD - PPCM

PRESENTATION

Ce programme détermine les Plus Grand Commun Dénominateur (PGCD) et Plus Petit Commun Multiple (PPCM) de 2 à 10 nombres entrés simultanément.

MODE D'EMPLOI

- 1) Indiquer le nombre de valeurs devant intervenir (entre 2 et 10 par INPUT).
- 2) Donner une à une chaque valeur (INPUT).
- 3) Affichage du PGCD et du PPCM de l'ensemble des valeurs entrées en 2.
- 4) Presser « * » pour continuer (INKEY\$).

FONCTIONS

CLS, DIM, INKEY\$, INT.

PROGRAMME

```

5 REM *** PGCD-PPCM ***
7 CLEAR :DIM A(10):A=1
10 BEEP 1:CLS :INPUT "CHOIX:2 A 10 NO
MBRES";B:IF B>=2 THEN IF B<=10 THEN CLS
:GOTO 20

```

```

15 GOTO 10
17 REM *ENTREES DES DONNEES*
20 FOR C=1 TO B:BEEP 1:PRINT "NOMBRE"
:C:INPUT D:IF D>1 THEN D=INT(D):A(C)=D
:GOTO 40
30 FOR G=1 TO 3:BEEP :NEXT G:GOTO 20
40 A=A*D:NEXT C:E=A(1):FOR C=2 TO B:F
=E:E=A(C)
45 REM *CALCUL DU PGCD ET DU PPCM*
50 U=-(INT(F/E)*E-F):IF U=0 THEN 70
60 F=E:E=U:GOTO 50
70 A=A/E:NEXT C:CLS :BEEP 1:PRINT "PG
CD ";"E,"PPCM ";"A
75 IF INKEY$<>"*" THEN 75
80 CLS :BEEP :PRINT "AUTRE CALCUL: 0/
N ?"
90 IF INKEY$=="0" THEN 7
100 IF INKEY$=="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR S=1 TO 300:NEXT S:END
110 GOTO 90

```

EXEMPLE

```

CHOIX:DE 2 A 10 NOMBRES?      NOMBRE 5?
5
NOMBRE 1?                      90
125
NOMBRE 2?                      PGCD= 5
375
NOMBRE 3?                      PPCM= 15552000
NOMBRE 4?                      *
UN AUTRE CALCUL:0/N?
400
NOMBRE 4?                      AU REVOIR,A BIENTOT!
720

```

MATHEMATIQUES

FACTORIELLES

PRESENTATION

Ce programme permet le calcul de factorielles de nombres supérieurs à 69 (limite de calcul du PB 700).

MODE D'EMPLOI

- 1) Entrer le nombre dont on désire calculer la grande factorielle (INPUT).
- 2) Affichage de la factorielle.
- 3) Presser « * » pour continuer (INKEY\$).

FONCTIONS

CLS, FRAC, INKEY\$, LGT, LOCATE.

PROGRAMME

```

5 REM *** GRANDES FACTORIELLES ***
10 CLS :BEEP 1:CLEAR :INPUT "VOTRE VA
LEUR ";N
15 REM *CALCUL*
20 FOR I=1 TO N:Z=Z+GT(I):NEXT I
25 REM *RESULTAT*
30 BEEP 1:CLS :PRINT "FACTORIELLE DE"
;N;"=":LOCATE 0,1

```

```
40 PRINT 10^(FRAC2); "*10 ^"; INT2
45 IF INKEY$<>"*" THEN 45
50 CLS :PRINT "UN AUTRE CALCUL:0/N?"
60 IF INKEY$="0" THEN 10
70 IF INKEY$="N" THEN PRINT "AU REVOIR,A BIENTOT!" :FOR G=1 TO 300:NEXT G:END
80 GOTO 60
```

EXEMPLE

```
VOTRE VALEUR ?
25
FACTORIELLE DE 25=
1.551121003*10 ^ 25
*
UN AUTRE CALCUL:0/N?
N
AU REVOIR,A BIENTOT!
```

VIE PRATIQUE

SIGNALISATION

PRESENTATION

La signalisation phonétique internationale, plus particulièrement utilisée en aviation, permet d'épeler des mots en codes connus par tous. Outre sa capacité de traduction, le programme suivant offre la possibilité d'un entraînement intensif par l'intermédiaire d'une fonction « EXERCICE ».

MODE D'EMPLOI

- 1) Choisir l'option (INKEY\$), soit :
T : pour traduction
E : pour exercices
F : pour fin.
- 2) Traduction : entrer un nombre de lettres ou de chiffres inférieur à 80 (INPUT).
... Affichage signe par signe de la traduction et retour par 1.
- 3) Exercice : traduction de 10 signes choisis aléatoirement par le programme (INPUT).
... Affichage du score et retour en 1.
- 4) Fin : pour arrêter le programme.

FONCTIONS

ASC, CLS, DATA, DIM, ELSE, INT, LEN, MID\$, READ, RND.

PROGRAMME

```
5 REM *** SIGNALISATION PHONETIQUE *
**
10 CLEAR :DATA ZERO,UN,UN ET UN,DEUX
ET UN,DEUX FOIS DEUX,TROIS ET DEUX
20 DATA DEUX FOIS TROIS,QUATRE ET TRO
IS,DEUX FOIS QUATRE,CINQ ET QUATRE
30 DATA ALPHA,BRAVO,CHARLIE,DELTA,ECH
O,FOX-TROT,GOLF,HOTEL,INDIA,JULIET
40 DATA KILO,LIMA,MIKE,NOUVEMBER,OHIO,P
APA,QUEBEC,ROMEO,SIERRA,TANGO
50 DATA UNIFORM,VICTOR,WHISKY,X-RAY,YA
NKEE,ZULU
60 DIM A$(36),C$(1),D$(1),N$(1)*29:FO
R J=1 TO 36:READ A$(J):NEXT J
70 CLS :BEEP 1:PRINT "***** OPTION *
***** (T)RADUC/(E)XOS/(F)?"
80 IF INKEY$="E" THEN 200
85 IF INKEY$="F" THEN CLS :END
90 IF INKEY$<>"T" THEN 80
100 BEEP 1:CLS :INPUT "VOS CARACTERES
";N$(1)
105 FOR Z=1 TO LEN(N$(1))
110 L=ASC(MID$(N$(1),Z,1)):X=0:IF L>47
THEN IF L<58 THEN 150
120 X=1:IF L>64 THEN IF L<91 THEN 150
140 C$(1)=" ":GOTO 160
150 IF X=0 THEN C$(1)=A$(L-47) ELSE C$
(1)=A$(L-54)
160 BEEP 1:IF Z=1 THEN CLS :PRINT "***"
* TRADUCTION ***
165 PRINT C$(1);"/";:FOR S=1 TO 150:NE
XT S:NEXT Z
170 IF INKEY$<>"*" THEN 170
180 GOTO 70
```

```
200 H=0:FOR G=1 TO 10:CLS :W=INT(RND*3
6+1):C$(1)=A$(W):IF W>10 THEN 250
210 BEEP 1:PRINT "TRADUCTION DE ";W-1::
INPUT D$(1)
220 IF D$(1)=C$(1) THEN BEEP 1:PRINT "
** BONNE REPONSE **":H=H+1:GOTO 240
230 PRINT "MAUVAISE REPONSE !!! C ETAIT
";C$(1):BEEP
240 FOR D=1 TO 200:NEXT D:NEXT G:FOR U
=1 TO 3:BEEP 1:BEEP 1:NEXT U:CLS
245 PRINT "**** TERMINE !! **** UOTRE S
CORE =";H*2;" / 20":GOTO 170
250 BEEP 1:PRINT "TRADUCTION DE ";LEFT
$(C$(1),1)::INPUT D$(1):GOTO 220
```

EXEMPLE

***** OPTION *****
(T)RADUC/(E)XOS/(F)?

T

VOS CARACTRERES ?

CASIO

CHARLIE/ALPHA/SIERRA
/INDIA/OHIO/

MORSE

PRESENTATION

Devenez un super-champion du code morse grâce au programme suivant qui, outre la traduction visuelle, offre une traduction sonore.

MODE D'EMPLOI

- 1) Choisir le signe d'espace qui séparera les différents codes en morse (INKEY\$).
 - 2) Sélectionner l'option (INKEY\$), soit :

T : pour la traduction alphabet morse
N : pour l'inverse
F : pour fin.

- 3) Traduction : entrer le message à traduire en lettres ou en codes morse et préciser par « O » (INKEY\$) si les codes morse devront être accompagnés de leur équivalent sonore.

FONCTIONS

ASC, CLS, DATA, DIM, ELSE, INKEY\$, LEN, MID\$, READ

```

190 CLS :INPUT "CODE MORSE ";C$(1):GOS
UB 330 .
195 REM *TRADUCTION MORSE-FRANCAIS*
200 H$(1)="( ) + -./0123456789      AB
CDEFGHIJKLMNOPQRSTUVWXYZ *,"
205 U=0:FOR P=1 TO LEN(C$(1)):IF MID$(C$(1),P,1)=S$ THEN U=P:GOTO 207
206 NEXT P:FOR P=1 TO 4:BEEP :NEXT P:G
OTO 185
207 IF U=1 THEN R=1:GOTO 206
208 Y=R+1:T$=MID$(C$(1),Y,U-Y):R=U:GOS
UB 370
210 FOR G=0 TO 54:IF T$<>A$(G) THEN NE
XT G:GOTO 240
220 L$=MID$(H$(1),G+1,1):GOSUB 250:IF
Y=2 THEN CLS :PRINT "**** TRADUCTION ***"
*
225 PRINT " ";L$;" ";K$(1);S$::GOTO 20
6
230 GOTO 185
240 BEEP 1:CLS :PRINT T$;"INCONNU!":G
OTO 185
245 REM *CAS PARTICULIERS*
250 K$(1)=""":U=ASC(L$):IF U=40 THEN K$(
1)="DEBUT"
260 IF U=41 THEN K$(1)="FIN ET COMPRIS
"
270 IF U=43 THEN K$(1)="R    RECU"
280 IF U=45 THEN K$(1)="Y    TRANSMETTE
Z"
290 IF U=46 THEN K$(1)="ATTENDEZ"
300 IF U=47 THEN K$(1)="ERREUR"
310 IF U=44 THEN K$(1)="REPETEZ"
320 RETURN
330 BEEP :PRINT "    SON : O/N ?"
340 IF INKEY$="O" THEN O=1:RETURN
350 IF INKEY$<>"N" THEN 340

```

```

360 O=0:RETURN
370 IF O=0 THEN RETURN
380 FOR D=1 TO LEN(T$):IF MID$(T$,D,1)
=".," THEN BEEP 1 ELSE BEEP
390 FOR X=1 TO 50:NEXT X:NEXT D:RETURN

```

EXEMPLE

SIGNE D ESPACE?
/
***** OPTION *****
(T)RADUC/(N)ORM/(F)?
T
VOS CARACTERES ?
CASIO
SON: O/N ?
N
**** TRADUCTION ****
-.-./.-/.../..-/--
***** OPTION *****
(T)RADUC/(N)ORM/(F)?
F
>

DATES

PRESENTATION

Programme multiple qui peut calculer les jours de Pâques, Ascension et Pentecôte, le nombre de jours entre deux dates ainsi que le jour d'une date.

MODE D'EMPLOI

1) Choisir l'option par INKEY\$, soit :

- 1 : pour dates de Pâques, Ascension et Pentecôte.
- 2 : pour le nombre de jours entre deux dates
- 3 : le jour d'une date
- F : pour fin.

- 2) L'option 1 demande l'année de calcul (INPUT) et affiche les dates des fêtes de l'année demandée.
- 3) L'option 2 demande les jours, les mois, les années des deux dates et affiche le nombre de jours les séparant.
- 4) L'option 3 demande le jour, le mois et l'année de la date dont on désire connaître le jour, puis l'affiche.
- 5) Dans tous les cas, on retourne au menu en pressant « * » par INKEY\$.

FONCTIONS

ABS, CLS, DATA, DIM, DRAW, FRAC, INKEY\$, INT, LEN, LOCATE, READ, RESTORE.

PROGRAMME

```

5 REM *** DATES ***
7 CLS :CLEAR :DIM A$(6)*8
10 BEEP 1:PRINT "***** OPTION *****"
(1) (2) (3) (F)"
20 D$=INKEY$: IF D$="3" THEN 280
25 IF D$="F" THEN CLS :END
30 IF D$="2" THEN 180
40 IF D$<>"1" THEN 20
45 REM *PAQUES,PENTECOTE,ASCENSION*
50 CLS :BEEP 1: INPUT "VOTRE ANNEE ";Z
:A=19*FRAC(Z/19):B=Z/100
55 CLS :BEEP :PRINT "*****";Z;" ***"
****":FOR W=1 TO 100:NEXT W
60 C=INTB:D=100*FRACB:E=C/4:F=INTC:G=
4*FRACE:H=INT((8+C)/25)
70 I=INT((1+C-H)/3): J=15+19*A+C-F-I:K
=INT(.5+30*FRAC(J/30))
80 L=D/4:M=INTL:N=4*FRACL:O=32+2*(G+M
)-K-N
90 P=INT(7*FRAC(O/7)+.5):Q=INT((A+11*
K+22*P)/451):R=(114+K+P-7*Q)/31
100 X=INTR:Y=INT(31*FRACR+1.5):CLS :BE
EP 1:PRINT "PAQUES      :";Y;" /";X
110 Y=Y+19:IF Y>31 THEN Y=Y-31:X=X+2:G
OTO 130
120 X=X+1
130 BEEP 1:PRINT "PENTECOTE :";Y;" /";
X
140 Y=Y-10:IF Y<=0 THEN Y=Y+31:X=X-1
150 BEEP 1:PRINT "ASCENSION :";Y;" /";
X
160 IF INKEY$<>"*" THEN 160
170 GOTO 7
175 REM *NB DE JOURS ENTRE 2 DATES*

```

```

180 CLS :BEEP 1:PRINT "***** 1ere DATE
*****":GOSUB 270
190 IF M=1 THEN GOSUB 250:GOTO 210
195 IF M=2 THEN GOSUB 250:GOTO 210
200 GOSUB 260
210 Z=N:CLS :BEEP 1:PRINT "***** 2eme
DATE ****":GOSUB 270
220 IF M=1 THEN GOSUB 250:GOTO 240
225 IF M=2 THEN GOSUB 250:GOTO 240
230 GOSUB 260
240 N=ABS(N-Z):BEEP :CLS :PRINT "NOMBR
E DE JOURS --> ";N:GOTO 160
250 N=365*A+INT((A-1)/4)+31*(M-1)-INT(
(A-1)/100)+INT((A-1)/400)+J:RETURN
260 N=365*A+INT(A/4)+31*(M-1)-INT((M-1
)*.4+2.7)-INT(A/100)+INT(A/400)+J:RETURN
270 INPUT "JOUR ";J;"MOIS ";M;"ANNEE "
:A:RETURN
275 REM *JOUR D UNE DATE*
280 DATA LUNDI,MARDI,MERCREDI,JEUDI,VEN
DREDI,SAMEDI,DIMANCHE
290 RESTORE 280:FOR U=0 TO 6:READ A$(U)
:NEXT U
300 CLS :BEEP 1:GOSUB 270:Q=A:W=M:IF M
-3>=0 THEN M=M+1:GOTO 320
310 M=13+M:A=A-1
320 N=INT(365.25*A)+INT(30.6*M)+J-6210
49:N=N-1:Y=(N-INT(N/7))*7
330 CLS :BEEP :PRINT "LE";J;" /";W;" /
";Q;"...":LOCATE 0,1
340 PRINT "EST UN ";A$(Y):DRAW(58,17)-
(LEN(A$(Y))*8+54,17):GOTO 160

```

EXEMPLE

***** OPTION *****	**** 2eme DATE ****
(1) (2) (3) (F)	TOUR?1
1	MOIS ? 12
VOITRE ANNEE ?	ANNEE ? 1983
1962	NOMBRE DE JOURS -->
***** 1962 *****	7925
PAQUES :	***** OPTION *****
PENTECOTE :	(1) (2) (3) (F)
ASCENSION :	3
*	JOUR ? 10
***** OPTION *****	MOIS ? 4
(1) (2) (3) (F)	ANNEE ? 1963
2	LE 10/4/1963.....
***** 1er DATE *****	EST UN MERCREDI
JOUR? 21	
MOIS ? 3	
ANNEE ? 1962	

AGENDA

PRESENTATION

Ce programme se charge de gérer un agenda téléphonique ayant les mêmes caractéristiques qu'une petite gestion de fichier.

MODE D'EMPLOI

Affichage du menu et choix de l'option (INKEY\$) :

- 1) Pour remettre à zéro le contenu des registres, deux options, soit :
(T) : pour initialisation totale.
(P) : pour initialisation partielle. Donner alors le numéro de la ligne à supprimer.
- 2) Permet d'entrer des messages de 1 à 63 caractères (**maximum** 20 phrases en version de base). Dès que toutes les phrases sont en mémoire, on retourne au menu en rentrant « , » (INPUT).
- 3) Option affichage, elle propose deux modes, soit :
(P) : pour un affichage partiel du contenu d'une lettre.
(E) : pour l'affichage de la totalité des messages en mémoire.

Choisir si l'affichage se fera sur écran ou sur imprimante (INKEY\$). Dans le cas de l'affichage sur écran, il convient de presser ENTER pour continuer.

FONCTIONS

ASC, CHR\$, CLS, DIM, INKEY\$, LPRINT.

PROGRAMME

```

3 REM *** AGENDA ***
5 GOTO 10
7 CLEAR :DIM A$(20)*63,G$(1)*63
10 CLS :Q=10:BEEP 1:PRINT "***** OPT
ION ***** (1) (2) (3) (S)"
20 U$=INKEY$: IF U$="S" THEN CLS :END
40 IF U$="3" THEN 200
50 IF U$="2" THEN 120
60 IF U$<>"1" THEN 20
70 CLS :BEEP 1:PRINT "** INITIALISATI
ON **(P)ARTIELLE/(T)OTALE"
80 IF INKEY$="T" THEN 7
90 IF INKEY$<>"P" THEN 80
100 CLS :BEEP 1:INPUT "NUMERO DE LIGNE
";N:IF N=0 THEN GOTO Q
105 IF N>0 THEN IF N<=I THEN 110
107 GOTO 100
110 A$(N)"":FOR J=N TO I:A$(J)=A$(J+1
):NEXT J:A$(I)"":I=I-1:GOTO 100
120 BEEP :CLS :I=I+1:IF I>20 THEN PRIN
T "** FICHIER SATURE **":GOTO 490
130 CLS :BEEP 1:PRINT I;:INPUT " : CAR
ACTERES ";A$(I):IF A$(I)=",," THEN I=I-1:
GOTO Q
140 IF I<>1 THEN GOSUB 500
150 GOTO 120
200 CLS :BEEP 1:PRINT "***** OPTION *
*****LISTE (E)NTIERE LISTE (P)ARTIEL
LE"

```

```

210 B=0:IF INKEY$="P" THEN B=1:GOTO 23
0
220 IF INKEY$<>"E" THEN 210
230 BEEP 1:CLS :PRINT "***** OPTION *
***** (I)MPRIMANTE          (E)CRAN"
240 L=0:IF INKEY$="I" THEN L=1:LPRINT
CHR$(27); "F3":GOTO 260
250 IF INKEY$<>"E" THEN 240
260 CLS :IF L=1 THEN GOSUB 400
265 IF B=1 THEN 280
270 FOR K=1 TO I:PRINT K;" : ";A$(K):B
EEP 1:IF L=1 THEN LPRINT K;" : ";A$(K):G
OTO 275
272 IF INKEY$<>CHR$(23) THEN 272
275 NEXT K:IF L=1 THEN LPRINT CHR$(27)
;"F4"
278 GOTO Q
280 CLS :BEEP 1:PRINT "**QUELLE LETTRE
?**"
290 U$=INKEY$:C=ASC(U$):IF C>64 THEN I
F C<91 THEN 300
295 GOTO 290
300 CLS :FOR K=1 TO I:IF MID$(A$(K),1,
1)<>CHR$(C) THEN 275
310 BEEP 1:PRINT K;" : ";A$(K):IF L=1
THEN LPRINT K;" : ";A$(K):GOTO 275
320 IF INKEY$<>CHR$(23) THEN 320
330 GOTO 275
400 CLS :BEEP 1:PRINT "***** OPTION *
***** COULEUR : DE 0 A 3 ?"
410 U$=INKEY$:D=ASC(U$):IF D>47 THEN I
F D<52 THEN D=D-48:GOTO 420
415 GOTO 410
420 LOCATE 0,1:PRINT "AFFICHAGE: (U) (
H) ?":BEEP 1
430 Y=0:IF INKEY$="U" THEN Y=1:GOTO 45
0

```

```

440 IF INKEY$<>"H" THEN 430
450 BEEP 1:LOCATE 0,1:PRINT "TAILLE :
DE 0 A 9 ?"
460 U$=INKEY$:T=ASC(U$):IF T>47 THEN I
F T<58 THEN 480
470 GOTO 460
480 T=T-48:LPRINT CHR$(27); "Y"; Y,CHR$(27);
"J"; D,CHR$(27); "S"; T:CLS :RETURN
490 FOR X=1 TO 200:NEXT X:I=I-1:GOTO Q
500 X=0:FOR U=1 TO I-1:IF A$(U)=<A$(U+1)
THEN 520
510 G$(1)=A$(U):A$(U)=A$(U+1):A$(U+1)=
G$(1):X=1
520 NEXT U:IF X=1 THEN 500
530 RETURN

```

EXEMPLE

1 : DUPONT 265.24.55
 2 : POIRIER 204.25.41
 3 : VALLART 547.14.56

CALENDRIER

PRESENTATION

Ce programme effectue les conversions du calendrier julien en calendrier grégorien et inversement.

MODE D'EMPLOI

1) Choisir l'option de conversion (INKEY\$), soit :

O : pour julien vers grégorien
N : pour grégorien vers julien

2) Entrer le nombre de jours juliens dans le premier cas ou la date de conversion dans le second. A noter que les codes des mois correspondent au barème suivant :

JANVIER	14	JUILLET	8
FEVRIER	15	AOUT	9
MARS	4	SEPTEMBRE	10
AVRIL	5	OCTOBRE	11
MAI	6	NOVEMBRE	12
JUIN	7	DECEMBRE	13

3) Affichage de la conversion, presser « 5 » pour continuer (INKEY\$).

FONCTIONS

CLS, ELSE, FRAC, INKEY\$, INT.

PROGRAMME

```

5 REM *** CONVERSION JULIEN <-> GREG
ORIEN ***
10 CLEAR :CLS :N=365.25:Q=30.6001
20 BEEP :PRINT "**** CONVERSION ****"
ULIEN---->GREGORIEN      O/N ?"
30 D$=INKEY$: IF D$="N" THEN 130
40 IF D$<>"O" THEN 30
50 CLS : INPUT "NOMBRE DE JOURS JULIEN
S ";J
55 REM *CALCUL DE LA DATE*
60 F=FRACJ+.5:IF JK2299161 THEN 80
70 Z=INT((J-1867216.25)/36245.25):A=J
+1+Z-INT(Z/4)
80 B=A+1524:C=INT((B-122.1)/N):D=INT(
N*C):E=INT((B-D)/Q)
90 IF E<=13 THEN M=E-1 ELSE M=E-13
100 J=B-D-INT(Q*E)+F
110 IF M<2 THEN A=C-4715 ELSE A=C-471
6
115 REM *RESULTAT*
120 BEEP 1:CLS :PRINT "DATE:";J;" / ";M;
"/";A:GOTO 260
130 CLS :C=1720994.5:BEEP :PRINT "DATE
<15/10/1582:O/N?"
140 Y$=INKEY$: IF Y$="O" THEN 200
150 IF Y$<>"N" THEN 140
155 REM *CALCUL DU NB DE JOURS JULIENS
*
160 GOSUB 240:Y=Y-7
170 IF M=14 THEN A=A-1
175 IF M=15 THEN A=A-1
180 J=INT(N*A)+INT(Q*M)+X+C-Y
185 REM *RESULTAT*

```

```

190 BEEP 1:CLS :PRINT "NOMBRE DE JOURS
JULIENS =";J:GOTO 260
200 GOSUB 240:IF A>=0 THEN 220
210 IF M=14 THEN A=A+1:GOTO 230
215 IF M=15 THEN A=A+1:GOTO 230
220 IF M=14 THEN A=A-1
225 IF M=15 THEN A=A-1
230 J=INT(N*A)+INT(Q*M)+X+C:GOTO 190
240 CLS :INPUT "JOUR ",X,"CODE DU MOIS
";M,"ANNEE ";A
250 INPUT "VALEUR DU SIECLE ";Y:X=X+.5
RETURN
260 IF INKEY$<>"*" THEN 260
270 BEEP :CLS :PRINT "UN AUTRE CALCUL:
O/N?"
280 IF INKEY$="O" THEN 10
290 IF INKEY$="N" THEN PRINT TAB(20);"
AU REVOIR,A BIENTOT!":FOR G=1 TO 300:NEX
T G:END
300 GOTO 280

```

EXEMPLE

```

**** CONVERSION ****
JULIEN--->GREGORIEN
O/N ?
N
DATE<15/10/1582:O/N?
N
JOUR ?
5
CODE DU MOIS
2
ANNEE ?
1983
VALEUR DU SIECLE ?
20
NOMBRE DE JOURS JULIENS
= 2445491 *
UN AUTRE CALCUL:O/N?
N
AU REVOIR,A BIENTOT!

```

ACRES-HECTARES

PRESENTATION

Très utilisées en agriculture, les mesures en acres ou en hectares seront très facilement converties par ce programme.

MODE D'EMPLOI

- 1) Choisir l'option de conversion (INKEY\$), soit :
 - A : pour acres
 - H : pour hectares
- 2) Dans les deux cas, indiquer la mesure à convertir (INPUT).
- 3) Affichage de la mesure convertie.
- 4) Presser « * » pour continuer.

FONCTIONS

CLS, INKEY\$, LOCATE.

PROGRAMME

```

5 REM *** CONVERSION ACRES-HECTARES
***
10 BEEP 1:PRINT "***** OPTION *****"
(A)CRE / (H)ECTARE ?"

```

```

20 G$=INKEY$: IF G$="H" THEN 80
30 IF G$<>"A" THEN 20
40 REM *ENTREES DES DONNEES ET RESULT
ATS*
50 CLS : INPUT "MESURE EN ACRES ";A
60 H=A/2.471:BEEP 1:CLS :PRINT "MESUR
E EN HECTARES =";H:GOTO 100
80 CLS : INPUT "MESURE EN HECTARES ";H
90 A=H*2.471:BEEP 1:CLS :PRINT "MESUR
E EN ACRES =":LOCATE 0,1:PRINT A
100 IF INKEY$<> "*" THEN 100
110 CLS :PRINT "UN AUTRE CALCUL:O/N?"
120 IF INKEY$="O" THEN 10
130 IF INKEY$="N" THEN PRINT "AU REVOI
R,A BIENTOT!":FOR R=1 TO 300:NEXT R:END
140 GOTO 120

```

EXEMPLE

```

***** OPTION *****
(A)CRE / (H)ECTARE ?
A
MESURE EN ACRES ?
2500
MESURE EN HECTARES =
1011.736139
*
UN AUTRE CALCUL:O/N?
N
AU REVOIR,A BIENTOT!

```

POIDS IDEAL

PRESENTATION

Pour garder une ligne de rêve, ce programme calcule votre poids idéal en tenant compte de votre taille, de votre âge et de votre sexe.

MODE D'EMPLOI

- 1) Entrer l'âge en années (INPUT).
- 2) Préciser la taille en centimètres (INPUT).
- 3) Indiquer, par l'initiale, le sexe concerné (INKEY\$).
- 4) Affichage du poids idéal (en kg).
- 5) Presser « * » pour continuer, et « O » pour retour en 1 (INKEY\$).

FONCTIONS

CLS, INKEY\$.

PROGRAMME

```

5 REM "*** POIDS IDEAL ***"
10 CLS :CLEAR :BEEP :PRINT "VOTRE AGE
";:INPUT A:IF A<10 THEN 10

```

```

20·BEEP :PRINT "VOTRE TAILLE EN CMS "
::INPUT B:IF B<140 THEN 20
30 BEEP :PRINT "(M)ALE/(F)EMELLE?"
40 C=0:IF INKEY$<>"M" THEN C=1:IF INK
EY$<>"F" THEN 40
50 CLS :PRINT "VOTRE POIDS IDEAL DO
IT ETRE:";B-110-C*7+A/20;"Kg"
60 IF INKEY$<>"*" THEN 60
70 BEEP :CLS :PRINT "UN AUTRE CALCUL:
D/N?"
80 IF INKEY$="O" THEN 10
90 IF INKEY$="N" THEN CLS :PRINT "AU
REVOIR...":FOR I=1 TO 300:NEXT I:CLS :EN
D
100 GOTO 80

```

EXEMPLE

VOTRE AGE?	VOTRE AGE?
21	24
VOTRE TAILLE EN CMS?	VOTRE TAILLE EN CMS?
175	182
(M)ALE/(F)EMELLE?	(M)ALE/(F)EMELLE?
F	M
VOTRE POIDS IDEAL DOIT ETRE: 59.05Kg	VOTRE POIDS IDEAL DOIT ETRE: 73.2Kg
*	*
UN AUTRE CALCUL:D/N?	UN AUTRE CALCUL:D/N?
O	N
	AU REVOIR...

VIE PRATIQUE

ESPERANCE DE VIE

PRESENTATION

Démonstration hors du commun de l'usage d'un micro-ordinateur de poche, ce programme calcule le nombre d'années que nous pouvons espérer vivre ou qui nous sont accordées avec sursis !

MODE D'EMPLOI

- 1) Répondre aux différentes questions du programme (INPUT et INKEY\$).
- 2) Affichage du verdict.
- 3) Presser « * » pour continuer.

FONCTIONS

ASC, CLS, ELSE, INKEY\$.

PROGRAMME

```

5 REM"***ESPERANCE DE VIE ***"
10 CLEAR :CLS :BEEP :PRINT "VOTRE AGE
";:INPUT A
20 BEEP :PRINT "(M)ALE/(F)EMELLE ?"
30 B=0:IF INKEY$<>"M" THEN B=3:IF INK
EY$<>"F" THEN 30

```

```

40 C=C+75+B: IF A>55 THEN C=C+5 ELSE I
F A>70 THEN C=C+8
50 BEEP :PRINT "FUMEZ-VOUS: O/N?"
60 IF INKEY$="O" THEN PRINT "COMBIEN
DE CIGARETTES PAR JOUR": INPUT B:C=C-B/1
0:GOTO 80
70 IF INKEY$<>"N" THEN 60
80 CLS :BEEP :PRINT "LIEU D,HABITATIO
N : (U)ILLE/(C)AMPAGNE?"
90 IF INKEY$="C" THEN C=C+2:GOTO 150
100 IF INKEY$<>"U" THEN 90
110 PRINT "A/ -10000 HABITANTS B/
1000
0-10000 HAB.";
120 PRINT "C/ +100000 HAB. ?"
125 IF INKEY$="" THEN 125
130 Z$=INKEY$: IF ASC(Z$)<65 THEN IF AS
C(Z$)>67 THEN 130
140 C=C-ASC(Z$)+62
150 CLS :BEEP :PRINT "AVEZ-VOUS UNE AU
TO.. O/N?"
160 B=-3: IF INKEY$<>"O" THEN B=2: IF IN
KEY$<>"N" THEN 160
170 C=C+B:BEEP :CLS :PRINT "FAITES-VOU
S DU SPORT O/N?"
180 B=-1: IF INKEY$<>"O" THEN B=3: IF IN
KEY$<>"N" THEN 180
190 C=C+B:BEEP :BEEP :BEEP 1:BEEP :BEE
P :BEEP 1:CLS :PRINT "**** VERDICT!!! ***
**";
200 IF A<=C THEN PRINT "VOUS ALLEZ VIU
RE... JUSQU,A";C;" ANS":GOTO 220
210 PRINT "... VOUS AVEZ EU UN SURSIS
DE";A-C;" ANNEES!";
220 IF INKEY$<>"*" THEN 220
230 CLS :BEEP :PRINT "UN AUTRE ESPOIR
O/N?"

```

```

240 IF INKEY$="O" THEN 10
250 IF INKEY$<>"N" THEN 240
260 CLS :BEEP :PRINT "AU REVOIR...":FO
R I=1 TO 300:NEXT I:CLS :BEEP :END

```

EXEMPLE

VOTRE AGE?
 25
 (M)ALE/(F)EMELLE ?
 M
 FUMEZ-VOUS: O/N?
 O
 COMBIEN DE CIGARETTES PAR JOUR?
 40
 LIEU D,HABITATION :
 (U)ILLE/(C)AMPAGNE?
 U
 A/ -10000 HABITANTS
 B/ 10000-100000 HAB.
 C/ +100000 HAB. ?
 B
 AVEZ-VOUS UNE AUTO..
 O/N?
 O
 FAITES-VOUS DU SPORT
 O/N?
 O
 **** VERDICT!!! ****
 VOUS ALLEZ VIURE...
 JUSQU,A 63 ANS
 *
 UN AUTRE ESPOIR O/N?
 N
 AU REVOIR...

MONNAIE

PRESENTATION

Particulièrement recommandé pour le versement de payes en espèces, « MONNAIE » permet de connaître avec exactitude le montant de chaque valeur en billets et en pièces de monnaie.

MODE D'EMPLOI

- 1) Indiquer si les résultats seront imprimés ou seulement affichés, soit « O » pour le premier cas, et « N » pour le second (INKEY\$).
- 2) Préciser la somme dont le programme déterminera l'équivalent dans chaque valeur (INPUT).
- 3) Affichage de « EN COURS... », le programme calcule. A la question « (S)UITE/(R)ESULTATS », presser « S » pour rentrer une nouvelle somme, ou « R » pour connaître le résultat final (INKEY\$).
- 4) Si « S » est pressé, le programme retourne en 2.
- 5) Si « R » est pressé, le programme donne le détail de la monnaie, et demande si l'on veut un nouveau cumul. Soit « O » pour retour en 1, et « N » pour arrêt.

FONCTIONS

CHR\$, CLS, DATA, DIM INKEY\$, LPRINT, READ, RESTORE.

PROGRAMME

```
5 REM "*** MONNAIE ***"
10 CLS :CLEAR :DIM A(12):DIM B(12):FOR I=0 TO 12:READ A(I):NEXT I:RESTORE
20 DATA 500,200,100,50,20,10,5,2,1,.5
.,2,.1,.05
30 BEEP :PRINT "IMPRIMANTE:O/N ?"
40 A=1:IF INKEY$<>"O" THEN A=0:IF INKEY$<>"N" THEN 40
45 REM "*** ENTREE DES SOMMES ***"
50 CLS :BEEP 1:C=C+1:PRINT "SOMME N";
C::INPUT D:F=F+D:CLS :PRINT "EN COURS...
"
60 IF D-A(E)>=0 THEN B(E)=B(E)+1:D=D-A(E):IF E<12 THEN 60 ELSE 80
70 IF E<12 THEN E=E+1:GOTO 60
80 BEEP :PRINT "(S)UITE/(R)ESULTATS"
90 IF INKEY$="S" THEN E=0:GOTO 50
100 IF INKEY$<>"R" THEN 90
105 REM "*** RESULTATS ***"
110 IF A=0 THEN CLS :FOR I=0 TO 12:PRINT B(I);" X";A(I):FOR J=1 TO 200:NEXT J:NEXT I:GOTO 160
120 LPRINT CHR$(28);CHR$(46)
130 LPRINT "SOMMES TOTALES";F
140 FOR I=0 TO 12:LPRINT B(I);" X";A(I):NEXT I:GOTO 160
150 FOR I=0 TO 12:LPRINT B(I);" X";A(I):NEXT I
160 CLS :PRINT "UN AUTRE CUMUL:O/N?"
170 IF INKEY$="N" THEN BEEP :CLS :PRINT "AU REVOIR...":FOR I=1 TO 300:NEXT I:END
180 IF INKEY$="O" THEN 10
190 GOTO 170
```

EXEMPLE

```
IMPRIMANTE:D/N ?  
0  
SOMME N 1 ?  
8650.35  
EN COURS...  
(S)UITE/(R)ESULTATS  
S  
SOMME N 2?  
4525.80  
EN COURS...  
(S)UITE/RESULTATS  
S  
SOMME N 3?  
6600.55  
EN COURS...  
(S)UITE/RESULTATS  
R  
  
SOMMES TOTALES 19776.7  
39 X 500  
0 X 200  
2 X 100  
1 X 50  
1 X 20  
0 X 10  
1 X 5  
0 X 2  
0 X 1  
2 X 0.5  
2 X 0.2  
2 X 0.1  
2 X 0.05  
  
UN AUTRE CUMUL:D/N ?  
N  
AU REVOIR...
```

VIE PRATIQUE

SURFACE CORPORELLE

PRESENTATION

Utilisant la formule de Dubois, ce programme calcule la surface du corps humain.

MODE D'EMPLOI

- 1) Entrer le poids en kilogrammes (INPUT).
- 2) Préciser la taille en mètres (INPUT).
- 3) Affichage de la surface du corps en mètres carrés.

FONCTIONS

CLS, INKEY\$, USING.

PROGRAMME

```
5 REM"*** SURFACE CORPORELLE ***"  
10 CLS :BEEP :PRINT "VOTRE POIDS (en  
Kg)":INPUT A  
20 BEEP :PRINT "VOTRE TAILLE (en M)":  
INPUT B  
30 CLS :BEEP 1:PRINT "VOTRE SURFACE C  
ORPORELLE EST EGALE A...";  
40 PRINT USING "#.####"; ". "; .202*A^.  
425*B^.725;" M2"
```

```
50 IF INKEY$<>"*" THEN 50
60 CLS :PRINT "UN AUTRE CALCUL:O/N?"
70 IF INKEY$="O" THEN 10
80 IF INKEY$<>"N" THEN 70
90 CLS :PRINT "AU REVOIR...":FOR I=1
TO 300:NEXT I:CLS :BEEP :END
```

EXEMPLE

```
VOTRE POIDS (en Kg)?
70
VOTRE TAILLE (en M)?1.70
VOTRE SURFACE CORPORELLE EST EGALE A...
...1.8050 M2
*
UN AUTRE CALCUL:O/N?
N
AU REVOIR...
```

UTILITAIRES

COPIE D'ECRAN

PRESENTATION

Ayant servi à la réalisation de nombreux exemples de cet ouvrage, la copie d'écran permet la recopie sur imprimante de la totalité de l'affichage.

MODE D'EMPLOI

Une fois lancé, le programme trace immédiatement l'écran tel qu'il se trouve. Pour copier l'écran en cours de programme, il suffit donc de rajouter GOSUB PROG N avec N correspondant à la zone où se trouve le programme de copie et de rajouter RETURN à la fin de la ligne 40 (voir TITRES).

FONCTIONS

CHR\$, ELSE, LPRINT, POINT.

PROGRAMME

```
5 REM *** COPIE D ECRAN ***
10 LPRINT CHR$(28);CHR$(37),"F2"
15 LPRINT "J0","D62,2,27,2,27,-162,62
,-162,62,2","J0"
20 BEEP :FOR L=0 TO 159:FOR H=0 TO 31
```

```
30 IF POINT(L,H)=1 THEN 50
40 NEXT H:NEXT L:BEEP 1:LPRINT "F8":I
F Z=0 THEN END ELSE RETURN
50 LPRINT "0";60-H;",";-L
60 LPRINT "A-.4,.4,.4,-.4"
70 FOR P=-.4 TO -.8 STEP -.2:LPRINT "
G1,.6,";P;"-.2":NEXT P:GOTO 40
```

EXEMPLE

♥TITRES SUR FA-10♥

UTILITAIRES

DACTYLOGRAPHIE

PRESENTATION

Apprentissage du clavier alphabétique, DACTYLOGRAPHIE est, en fait, un jeu qui a des chances d'occuper vos longues soirées d'hiver.

MODE D'EMPLOI

- 1) Choisir l'option de jeu (INKEY\$), soit :
 - 1 : pour mot
 - 2 : pour phrase.
- 2) Le programme affiche soit un mot, soit une phrase, pris au hasard dans sa bibliothèque.
- 3) Il s'agit de presser le plus vite possible les touches correspondant aux lettres des mots affichés (INKEY\$).
- 4) Dès que la lettre juste est pressée, elle s'efface de l'écran et ainsi de suite jusqu'à la fin du mot ou de la phrase.
- 5) Affichage du pourcentage de réussite.

FONCTIONS

CHR\$, CLS, DATA, DIM, ELSE, INKEY\$, INT, LOCATE,
MID\$, READ, RND, VAL.

PROGRAMME

```
5 REM *** DACTYLOGRAPHIE ***
7 CLS :CLEAR :DIM A$(20)*40
8 REM *MOTS ET PHRASES*
10 DATA PHYSIOLOGIE,HYSTERIE,AUSTRALOP
    ITHEQUE,BANQUEROUTE
20 DATA FONDATION,HEMATOLOGIE,HELICOID
    AL,WALLONISME
30 DATA TRYPARASOME,INDUCTANCE,LE CHAT
    EST SORTI DEHORS
40 DATA UNE BOMBE A EXPLOSE AU CARREFO
    UR,LE NIVEAU D'EAU EST ELEVE
50 DATA LE LEZARD EST MORT,L AVION S
    EST ECRASE DANS LA FORET
60 DATA ELLE EST PARTIE A LA PECHE,LES
    ENNEMIS PERDENT DU TERRAIN
70 DATA UNE COULEE DE LAVE APPROCHE,LA
    LUNE EST NOTRE SATELLITE
80 DATA LA PANTHERE A DEVORE UN LIONCE
    AU
85 FOR S=1 TO 20:READ A$(S):NEXT S
90 BEEP 1:E=0:PRINT "***** OPTION **"
***** (1) MOT / (2) PHRASE"
100 U$=INKEY$:U=VAL(U$):IF U>0 THEN IF
    U<3 THEN 120
110 GOTO 100
115 REM *CHOIX DU MOT ET JEUX*
120 B=INT(RND*10+1)+(U-1)*10:FOR T=1 T
    O 3:BEEP :BEEP 1:BEEP :NEXT T:CLS
130 PRINT "** C EST PARTI !! **":FOR X
    =1 TO 100:NEXT X
140 LOCATE 0,1:PRINT A$(B):BEEP
150 FOR N=1 TO LEN(A$(B))
155 IF INKEY$=MID$(A$(B),N,1) THEN 250
160 E=E+1:IF E>100 THEN 260 ELSE 155
```

```
180 CLS :FOR L=1 TO 4:BEEP 1:BEEP :BEE
    P 1:NEXT L
190 PRINT "***** BRAVO !! *****":LOCAT
    E 0,1:PRINT 100-INTE;" ";CHR$(37);
195 PRINT " DE REUSSITE !"
200 BEEP :PRINT "REJOUEZ-VOUS: O/N ?"
210 IF INKEY$="O" THEN 90
220 IF INKEY$="N" THEN CLS :PRINT "AU
    REVOIR,A BIENTOT!":FOR Z=1 TO 300:NEXT Z
    :END
230 GOTO 210
240 REM *TOUCHE*
250 U=1:D=N:IF N>20 THEN U=2:D=N-20
255 LOCATE D-1,U:PRINT " ":"BEEP 1:NEXT
    N:GOTO 180
260 CLS :FOR C=1 TO 3:BEEP :NEXT C:PRI
    NT "***** PERDU !! *****":GOTO 200
```

EXEMPLE

***** OPTION *****	***** BRAVO !!*****
(1) MOT / (2) PHRASE	66 % DE REUSSITE !
1	REJOUEZ-VOUS: O/N ?
** C EST PARTIE !! **	N
HYSSTERIE	AU REVOIR,A BIENTOT!
STERIE	
TERIE	
ERIE	
RIE	
IE	
E	

MACHINE A Ecrire

PRESENTATION

Mini-traitement de texte, ce programme permet d'utiliser le PB 700 muni de son imprimante comme une machine à écrire.

MODE D'EMPLOI

- 1) Dès le départ, le programme fonctionne comme une machine à écrire. Les caractères saisis seront donc directement imprimés.
- 2) Pour modifier les caractéristiques d'impression appuyer sur « CLS » par INKEY\$.

Cette procédure permet d'accéder aux détails suivants :

Majuscules/minuscules,
 Couleur d'impression (1 pour noir, 2 pour bleu,
 3 pour vert et 4 pour rouge),
 Taille du caractère de 0 à 9,
 Espacement entre les caractères de 0 à 9,
 Interligne de 1 à 9.

- 3) Fonctions spéciales :
 ENTER fait avancer le papier d'une ligne.
 DEL provoque le retour du chariot
 → avance de 5 lignes et arrête le programme.

FONCTIONS

ASC, CHR\$, CLS, INKEY\$, LPRINT, VAL.

PROGRAMME

```

5 REM *** MACHINE A ECRIRE ***
7 CLS :CLEAR
10 LPRINT CHR$(28);CHR$(46):BEEP :X=1
2
15 PRINT "*VOUS POUVEZ ECRIRE*":FOR U
=1 TO 200:NEXT U:BEEP :BEEP 1:CLS :GOTO
20
17 LPRINT CHR$(27)::RETURN
20 G$=INKEY$: IF G$=CHR$(12) THEN GOSU
B 100
30 REM *DEROULEMENT*
40 IF G$=CHR$(28) THEN GOSUB X:LPRINT
"F4":END
50 IF G$=CHR$(23) THEN GOSUB X:LPRINT
"F1"
55 REM *RETOUR CHARIOT*
60 IF G$=CHR$(17) THEN G$=CHR$(23):GO
SUB X:LPRINT "T0":GOTO 50
65 REM *IMPRESSION*
70 GOSUB X:LPRINT "S":S:GOSUB X:LPRIN
T "J":J:GOSUB X:LPRINT "Z";Y;";Z
71 L=ASC(G$):IF L>64 THEN IF L<91 THE
N 80
72 IF L=0 THEN 20
75 LPRINT G$::GOTO 20
80 LPRINT CHR$(L+M)::GOTO 20
90 REM *CORRECTIONS*
100 BEEP 1:CLS :PRINT "***** OPTION *
***** (1): MAJUSCULES (2): MINUSCUL
ES"
105 U$=INKEY$:C=VAL(U$):IF C>0 THEN IF
C<3 THEN 110
107 GOTO 105
110 M=0:IF C=2 THEN M=32

```

```
120 CLS :BEEP 1:PRINT "COULEUR : DE 1  
A 4 ?"  
125 U$=INKEY$: J=VAL(U$): IF J>0 THEN IF  
J<5 THEN J=J-1:GOTO 130  
127 GOTO 125  
130 BEEP 1:CLS :PRINT "TAILLE : DE 0 A  
9 ?"  
135 U$=INKEY$: IF ASC(U$)>47 THEN IF AS  
C(U$)<58 THEN S=ASC(U$)-48:GOTO 140  
137 GOTO 135  
140 BEEP 1:CLS :PRINT "ESPACE : DE 1 A  
9 ?"  
145 U$=INKEY$: Y=VAL(U$): IF Y>0 THEN IF  
Y<10 THEN 150  
147 GOTO 145  
150 CLS :BEEP 1:PRINT "INTERLIGNE:DE 1  
A 9?"  
155 U$=INKEY$: Z=VAL(U$): IF Z>0 THEN IF  
Z<10 THEN CLS :BEEP :G$="":RETURN  
157 GOTO 155
```

EXEMPLE

J, AIME
C A S I O

UTILITAIRES

TITRES

PRESENTATION

Permettant de réaliser en copie d'écran une série de titres d'un nombre maximum de 18 signes.

MODE D'EMPLOI

- 1) Charger le programme **Titres** en P0 et le programme **Copie d'écran** en P1.
- 2) Rajouter RETURN à la fin de la ligne 40 du programme **Copie d'écran**.
- 3) Lancer le programme **Titres** par « SHIFT » « 0 ».
- 4) Entrer le nombre de titres dont on désire l'impression (INPUT).
- 5) Préciser, un à un, les titres devant être imprimés. Maximum 60 titres de 18 caractères.
- 6) Impression automatique de l'ensemble des titres.

FONCTIONS

CHR\$, CLS, DIM, INT, LEN, LOCATE, PROG, RND.

PROGRAMME

```
10 CLEAR :DIM A$(100)*18:DIM B$(1)*60
20 CLS :LOCATE 1,1:PRINT "NOMBRE DE T
ITRES";:INPUT A:IF A>60 THEN 20
30 FOR B=1 TO A
40 CLS :LOCATE 6,1:PRINT "TITRE N";:P
RINT B:INPUT B$(1)
50 IF LEN(B$(1))>18 THEN 40
60 A$(B)=B$(1):NEXT B
70 FOR B=1 TO A:C=INT(RND*4):A$(B)=CH
R$(232+C)+A$(B)+CHR$(232+C)
120 CLS :LOCATE ((20-LEN(A$(B)))/2),1:
PRINT A$(B):GOSUB PROG 1
130 NEXT B:END
```

COLLECTION E.T.S.F. MICRO-SYSTEMES

- 1 - A. VILLARD et M. MIAUX, *Un microprocesseur pas à pas*
- 2 - A. VILLARD et M. MIAUX, *Systèmes à microprocesseur*
- 3 - P. GUEULLE, *Maîtrisez votre ZX 81*
- 4 - E. FLOEGEL, *Du Basic au Pascal*
- 5 - P. COURBIER, *Vous avez dit Basic ?*
- 6 - M. MARCHAND, *Vous avez dit micro ?*
- 7 - P. GUEULLE, *Pilotez votre ZX 81*
- 8 - M. JACQUELIN, *La micro-informatique et son A-B-C*
- 9 - M. OURY, *Maîtrisez les TO 7 et TO 7-70*
- 10 - P. GUEULLE, *Pilotez votre Oric 1 + ATMOS*
- 11 - P. JOUVELOT et D. LE CONTE DES FLORIS, *Système d'exploita-
tion et logiciel de base des micro-ordinateurs*
- 12 - P. GUEULLE, *Robotisez votre ZX 81*
- 13 - M. CAUT, *J'apprends le Basic*
- 14 - C. MALOSSE, C. TASSET, P. PRUT, *La micro, c'est pas sorcier !*
- 15 - R. GREGOIRE, *Bus IEEE*
- 16 - M. OURY, *Maîtrisez le MO5*
- 17 - P. GUEULLE, *Votre ordinateur et la télématique*

Achevé d'imprimer
sur les presses
de l'Imprimerie Marcel Bon
70001 Vesoul
Dépôt légal : septembre 1984
Nº d'éditeur : 431
Nº d'imprimeur : 2825

40 PROGRAMMES POUR CASIO PB 700

Ce recueil illustre, par des **applications utiles ou amusantes**, les nombreuses fonctions du BASIC CASIO sur PB 700. Chaque programme, accompagné d'un exemple, est immédiatement utilisable.

Vous ferez ainsi, de façon distrayante, le tour des possibilités de cette machine et de son étonnante **imprimante traçante**, indispensable pour les programmes de graphisme.

Quelques programmes :

- Super mind
- Logic 100
- Grapho mémoire
- Cercles et ellipses
- Graphismes divers
- Volumes
- Grandes factorielles
- Morse
- Copie d'écran
- Dactylographie